



TECHNISCHE UNIVERSITÄT BERLIN  
FAKULTÄT IV – ELEKTROTECHNIK UND INFORMATIK  
MODELLE UND THEORIE VERTEILTER SYSTEME

BACHELOR'S THESIS

# Reducing Strong Reactive Bisimilarity to Strong Bisimilarity

MAXIMILIAN POHLMANN  
STUDENT ID: 387210

FIRST EVALUATOR: PROF. DR. UWE NESTMANN  
SECOND EVALUATOR: PROF. DR. STEPHAN KREUTZER

BERLIN  
JUNE 2021



# Contents

<b>Contents</b>	<b>3</b>
<b>1 Introduction</b>	<b>5</b>
<b>2 Foundations</b>	<b>8</b>
2.1 Labelled Transition Systems . . . . .	8
2.2 Strong Bisimilarity . . . . .	11
2.3 Hennessy-Milner Logic . . . . .	13
2.4 Labelled Transition Systems with Time-Outs . . . . .	20
2.5 Reactive Bisimilarity . . . . .	24
2.6 Hennessy-Milner Logic with Time-Outs . . . . .	30
<b>3 The Reductions</b>	<b>33</b>
3.1 A Mapping for Transition Systems . . . . .	34
3.2 Reduction of Bisimilarity . . . . .	39
3.3 A Mapping for Formulas . . . . .	41
3.4 Reduction of Formula Satisfaction . . . . .	43
<b>4 Discussion</b>	<b>44</b>
<b>Bibliography</b>	<b>47</b>
<b>A Introduction to Isabelle</b>	<b>49</b>
<b>B Infinitary Hennessy-Milner Logic</b>	<b>53</b>
<b>C Example Instantiation</b>	<b>57</b>



# Chapter 1

## Introduction

In this thesis, I show that it is possible to reduce the problem of checking strong reactive bisimilarity, as introduced by Rob van Glabbeek in [vG20], to checking ordinary strong bisimilarity. I do this by specifying a mapping that yields a model of the closed system consisting of some original (reactive) system and its environment. I formalised all concepts discussed in this thesis, and conducted all the proofs, in the interactive proof assistant Isabelle.

Reactive systems are systems that continuously interact with their environment (e.g. a user) and whose behaviour is largely dependent on this interaction [HP85]. They can be modelled using labelled transition systems (LTSs) [Kel76]; roughly, an LTS is a labelled directed graph, whose nodes correspond to states of a reactive system and whose edges correspond to transitions between those states.<sup>1</sup>

A user interacting with some system can only perceive it in terms of the interactions it reacts to, i.e. the internal state of the system is hidden from the user. This begets a notion of behavioural/observational equivalence: two non-identical systems may exhibit equivalent behaviour as observed by the user. The simplest such equivalence is known as *strong bisimilarity*.

In classical LTSs, a system cannot react to the absence of interaction, as it would be assumed to simply wait for any interaction. Intuitively, however, a system may be equipped with a clock and perform some activity when it has seen no interaction from the user for a specified time. Such a system would not be describable with classical LTS semantics. Amongst these systems are, e.g., systems implementing mutual exclusion protocols [vG20].

---

<sup>1</sup>The topics of this thesis are applicable to any such graphs in an abstract way. However, I will continue to use motivations and terminology derived from the interpretation of LTSs as reactive systems.

In [vG21], Rob van Glabbeek introduces labelled transition systems with time-outs ( $\text{LTS}_t$ )<sup>2</sup>, which allow for modelling such systems as well. The appertaining equivalence is given in [vG20] as *strong reactive bisimilarity*.

For the first main result of this thesis, I show that it is possible to reduce checking strong reactive bisimilarity to checking strong bisimilarity. This is in line with reductions of other behavioural equivalences to strong bisimilarity. For example, a strategy used to reduce *weak bisimilarity* to strong bisimilarity is called *saturation* and is described in [AIS11, section 3.2.5].

The strategy used for reducing reactive bisimilarity to strong bisimilarity is based on the fact that the concept of strong reactive bisimilarity requires an explicit consideration of the environments in which specified systems may exist. Concretely, I specify a mapping from  $\text{LTS}_t$ s to LTSs, where each state of the mapped LTS corresponds to a state of the original  $\text{LTS}_t$  in some specific environment. This approach has been hinted at by van Glabbeek at various points in [vG20], but has not been made explicit.

The reduction of reactive bisimilarity could be of use in the context of automated model checking tools: there are known algorithms for checking equivalences (e.g. see [AIS11]) and tools with efficient implementations thereof;<sup>3</sup> instead of implementing an algorithm for checking strong reactive bisimilarity from scratch, an implementation of the reduction would allow the use of these existing implementations. Moreover, the mapping used for the reduction may aid in the analysis of system specifications using  $\text{LTS}_t$ s, by providing a more explicit view at the system.

Another interesting way to examine the behaviour of an LTS is through the use of modal logics, where formulas describe certain properties and are evaluated on states of an LTS. A commonly used modal logic is known as Hennessy-Milner logic (HML). An extension of HML for evaluation on states of an  $\text{LTS}_t$  is also given in [vG20]; I will refer to this extension as Hennessy-Milner logic with time-outs ( $\text{HML}_t$ ).

For the second main result of this thesis, I show that it is possible to reduce formula satisfaction of  $\text{HML}_t$  on  $\text{LTS}_t$ s to formula satisfaction of HML on LTSs (using another mapping for formulas, along with the mapping from the first reduction).

---

<sup>2</sup>He does not use that specific term or abbreviation, however.

<sup>3</sup>e.g. see LTSmin at [github.com/utwente-fmt/ltsmin](https://github.com/utwente-fmt/ltsmin)

## How This Thesis is Structured / Isabelle

The remainder of this thesis is split into [Foundations](#) ([chapter 2](#)), where LTSs, bisimilarity, and Hennessy-Milner logic, all without and with time-outs, are discussed and formalised, and [The Reductions](#) ([chapter 3](#)), where the reduction of bisimilarity and the reduction of formula satisfaction are presented in detail and proved.

All the main topics of this thesis have been formalised, and all proofs conducted, using the interactive proof assistant Isabelle. More information on Isabelle and a short introduction to the most important concepts can be found in [appendix A](#).

This thesis document itself was generated using the Isabelle document preparation system (see [\[Wen21a\]](#)), which generates L<sup>A</sup>T<sub>E</sub>X markup from Isabelle code (and, of course, integrates markup written manually). This allowed me to integrate all the Isabelle code directly into the thesis document. However, almost all proofs are hidden (and replaced simply by *<proof>*) and some lemmas excluded. In these cases, an indication of the proof strategy used is given in text. A web version of this thesis, that includes all formalisations, propositions, and proofs, as well as all the text, can be found on GitHub Pages, with one page for each section of this thesis.<sup>4</sup>

All of the sections of [chapters 2](#) and [3](#) are split into two parts: one containing a prosaic and mathematical description of the topics, and one containing the (documented) formalisation/implementation in Isabelle. I try to clearly distinguish between mathematical structures and their implementation. Although the two are, necessarily, closely related, they are not identical. The former is written in L<sup>A</sup>T<sub>E</sub>X math mode in this *italic font*, the latter is Isabelle code in this `monospaced font`.

---

<sup>4</sup>see [maxpohlmann.github.io/Reducing-Reactive-to-Strong-Bisimilarity](https://maxpohlmann.github.io/Reducing-Reactive-to-Strong-Bisimilarity)

## Chapter 2

# Foundations

In this chapter, the concepts that are relevant for the main part of this thesis will be introduced in text, as well as formalised in Isabelle. The formalisations of [sections 2.1](#) and [2.2](#) are based on those done by Benjamin Bisping in [\[BN19\]](#) (the code is available on GitHub<sup>1</sup>). All other formalisations were done as part of this thesis.

### 2.1 Labelled Transition Systems

---

A Labelled Transition System (LTS) consists of a set of processes (or states)  $Proc$ , a set of actions  $Act$ , and a relation of transitions  $\cdot \xrightarrow{\cdot} \cdot \subseteq Proc \times Act \times Proc$  which directly connect two processes by an action (the action being the label of the transition) [\[AILS07\]](#). We call a transition labelled by an action  $\alpha$  an  $\alpha$ -transition.

LTSs can model reactive systems, as discussed in [chapter 1](#). A process of an LTS, then, corresponds to a momentary state of a reactive system. The outgoing transitions of each process correspond to the actions the reactive system could perform in that state (yielding a subsequent process/state), if facilitated by the environment. The choice between the facilitated transitions of a process is non-deterministic.

This facilitation can be thought of as a set of actions that the environment allows in a given moment, or, more intuitively, as a set of simultaneous inputs from the environment to which the system may react. We call the actions not allowed by the environment in a given moment *blocked*.

The environment can also observe which transitions a system performs and react by changing the set of allowed actions in response.


---

<sup>1</sup>see [coupledsim.bbisping.de/code/](https://coupledsim.bbisping.de/code/)



In classical treatments of LTSs, the actions that the environment allows, blocks, or reacts to, are often considered only implicitly. The reason we put this emphasis on the environment here will become apparent in [section 2.4](#).

**Example** The process  $p$  can perform any of the  $a$ -transitions in environments allowing  $a$  and the  $b$ -transition in environments allowing  $b$ . All derivative (subsequent) states cannot perform any transition.



A *hidden action*, denoted by  $\tau$ , allows for additional semantics: a  $\tau$ -transition can be performed by a process without any interaction from the environment. Depending on the specific semantic context, the performance of a hidden action may also be unobservable (hence the name), i.e. not part of the observable behaviour of the system.

### Some Definitions

The  $\alpha$ -derivatives of a state are those states that can be reached with one  $\alpha$ -transition:

$$Der(p, \alpha) = \{p' \mid p \xrightarrow{\alpha} p'\}.$$

An LTS is image-finite iff all derivative sets are finite, i.e.:

$$\forall p \in Proc, \alpha \in Act. Der(p, \alpha) \text{ is finite.}$$

Similarly, we can say that an LTS is image-countable iff all derivative sets are countable, i.e.:

$$\forall p \in Proc, \alpha \in Act. Der(p, \alpha) \text{ is countable.}$$


### Note on Metavariable usage

States of an LTS range over  $p, q, p', q', \dots$ , where  $p$  and  $p'$  are used for states connected by some transition (i.e.  $p \xrightarrow{\alpha} p'$ ), whereas  $p$  and  $q$  are used for states possibly related by some equivalence (e.g.  $p \leftrightarrow q$ ), as will be discussed in the next section.

An arbitrary action of an LTS will be referenced by  $\alpha$ , whereas an arbitrary *visible* (i.e. non-hidden) action will be referenced by  $a$ .

### Practical Example

Let us take a detour away from purely theoretical deliberations and consider a real-world machine that may be modelled as an LTS. We can imagine a very simple snack-selling vending machine that accepts only one type of coin and has individual buttons for each of the snacks. When a coin is inserted and a button pressed, the machine dispenses the desired snack and is then ready to accept coins once again. Because the dispensing of the snack itself requires no interaction from the user, we model it as a  $\tau$ -transition.




---

Isabelle

---

The sets of states and actions are formalised by type variables 's and 'a, respectively. A specific LTS on these sets is then determined entirely by its set of transitions, denoted by the predicate `tran`. We associate it with a more readable notation ( $p \mapsto_{\alpha} p'$  for  $p \xrightarrow{\alpha} p'$ ).

```

locale lts =
  fixes tran :: ('s  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool)
    ( $\_ \mapsto_{\_}$  [70, 70, 70] 80)
  begin

```

The other definitions can be formalised in a straightforward manner.

```

abbreviation derivatives :: ('s  $\Rightarrow$  'a  $\Rightarrow$  's set)
  ( $\langle$ Der'(_, _)'  $\rangle$  [50, 50] 1000)
  where  $\langle$ Der(p,  $\alpha$ )  $\equiv$  {p'. p  $\mapsto_{\alpha}$  p'} $\rangle$ 

```

```

definition image_finite :: (bool)
  where  $\langle$ image_finite  $\equiv$  ( $\forall$  p  $\alpha$ . finite Der(p,  $\alpha$ )) $\rangle$ 
definition image_countable :: (bool)
  where  $\langle$ image_countable  $\equiv$  ( $\forall$  p  $\alpha$ . countable Der(p,  $\alpha$ )) $\rangle$ 

```

These two properties concern the entire LTS at hand (given by the locale context) and will be useful when we want to state propositions that only hold for LTSs that satisfy these properties.

```

end — of locale lts

```

We formalise LTSs with hidden actions as an extension of ordinary LTSs with a fixed action  $\tau$ .

```

locale lts_tau = lts tran
  for tran :: ⟨'s ⇒ 'a ⇒ 's ⇒ bool⟩ +
  fixes  $\tau$  :: ⟨'a⟩

```


## 2.2 Strong Bisimilarity

---

As discussed in the previous section, LTSs can describe the behaviour of reactive systems, and this behaviour is observable by the environment (in terms of the transitions performed by the system). This begets a notion of behavioural equivalence, where two processes are said to be behaviourally equivalent if they exhibit the same (observable) behaviour [AILS07].


Bisimilarity (or *strong bisimilarity*, to be precise) is the ‘*finest extensional behavioural equivalence* [...] on processes’ [San12, section 0.1], an extensional property being one that treats the system in question as a black box, i.e. the specific state space of the system remains hidden and performed transitions are only observable in terms of their action-label. This distinguishes bisimilarity from stronger graph equivalences like *graph isomorphism*, where the (intensional) identity of processes (graph nodes) is relevant [San11].

**Example** The processes  $p$  and  $q$  are strongly bisimilar (written  $p \leftrightarrow q$ , following [vG20]), as both can always perform exactly two  $a$ -transitions and no further transitions afterwards. There is no isomorphism between the left and right subgraphs, as they have a different number of nodes.



It is important to note that not only transitions that are performable, but also those that are not, are relevant.

**Example** The processes  $p$  and  $q$  are not strongly bisimilar, as  $p$  can perform an  $a$ -transition into a subsequent state, where it can perform no further transitions, whereas  $q$  can always perform two  $a$ -transitions in sequence.



Strong bisimilarity is the *finest* extensional behavioural equivalence, because all actions are thought of as observable. An example of a coarser equivalence is *weak bisimilarity*, which treats the aforementioned hidden action  $\tau$  as unobservable. However, weak bisimilarity is of no further relevance for this thesis and the interested reader is referred to [AILS07, Chapter 3.4].

The notion of strong bisimilarity can be formalised through *strong bisimulation* (SB) relations, introduced originally by David Park in [Par81]. A binary relation  $\mathcal{R}$  over the set of processes  $Proc$  is an SB iff for all  $(p, q) \in \mathcal{R}$ :

$$\begin{aligned} \forall p' \in Proc, \alpha \in Act. p \xrightarrow{\alpha} p' &\longrightarrow \exists q' \in Proc. q \xrightarrow{\alpha} q' \wedge (p', q') \in \mathcal{R}, \text{ and} \\ \forall q' \in Proc, \alpha \in Act. q \xrightarrow{\alpha} q' &\longrightarrow \exists p' \in Proc. p \xrightarrow{\alpha} p' \wedge (p', q') \in \mathcal{R}. \end{aligned}$$

---

Isabelle

---

Strong bisimulations are straightforward to formalise in Isabelle, using the ‘curried’ definition approach discussed in [appendix A](#).

**context lts begin**

— strong bisimulation

**definition** SB ::  $\langle ('s \Rightarrow 's \Rightarrow \text{bool}) \Rightarrow \text{bool} \rangle$

**where**  $\langle \text{SB } R \equiv \forall p \ q. R \ p \ q \longrightarrow$

$$\begin{aligned} &(\forall p' \ \alpha. p \xrightarrow{\alpha} p' \longrightarrow (\exists q'. (q \xrightarrow{\alpha} q') \wedge R \ p' \ q')) \wedge \\ &(\forall q' \ \alpha. q \xrightarrow{\alpha} q' \longrightarrow (\exists p'. (p \xrightarrow{\alpha} p') \wedge R \ p' \ q')) \rangle \end{aligned}$$

Two processes  $p$  and  $q$  are then strongly bisimilar iff there is an SB that contains the pair  $(p, q)$ .

**definition** `strongly_bisimilar` ::  $\langle 's \Rightarrow 's \Rightarrow \text{bool} \rangle$   
 $\langle \_ \leftrightarrow \_ \rangle [70, 70] 70$   
**where**  $\langle p \leftrightarrow q \equiv \exists R. \text{SB } R \wedge R \ p \ q \rangle$

The following corollaries are immediate consequences of these definitions.

**corollary** `strongly_bisimilar_step`:

**assumes**  $\langle p \leftrightarrow q \rangle$   
**shows**  
 $\langle p \xrightarrow{a} p' \implies (\exists q'. (q \xrightarrow{a} q') \wedge p' \leftrightarrow q') \rangle$   
 $\langle q \xrightarrow{a} q' \implies (\exists p'. (p \xrightarrow{a} p') \wedge p' \leftrightarrow q') \rangle$   
 $\langle \text{proof} \rangle$

**corollary** `strongly_bisimilar_symm`:

**assumes**  $\langle p \leftrightarrow q \rangle$   
**shows**  $\langle q \leftrightarrow p \rangle$   
 $\langle \text{proof} \rangle$

**end** — context `lts`

## 2.3 Hennessy-Milner Logic

---

In their seminal paper [HM85], Matthew Hennessy and Robin Milner present a modal-logical characterisation of strong bisimilarity (although they do not call it that), by process properties: ‘two processes are equivalent if and only if they enjoy the same set of properties.’ These properties are expressed as terms of a modal-logical language, consisting merely of (finite) conjunction, negation, and a family of modal possibility operators. This language is known today as Hennessy-Milner logic (HML), with formulas  $\varphi$  defined by the following grammar (where  $\alpha$  ranges over the set of actions  $Act$ ):

$$\varphi ::= \# \mid \varphi_1 \wedge \varphi_2 \mid \neg \varphi \mid \langle \alpha \rangle \varphi$$

The semantics (on LTS processes) is given as follows: all processes satisfy  $\#$ ,  $\varphi_1 \wedge \varphi_2$  is satisfied if both  $\varphi_1$  and  $\varphi_2$  are satisfied,  $\neg \varphi$  is satisfied if  $\varphi$  is not satisfied, and  $\langle \alpha \rangle \varphi$  is satisfied by a process if it is possible to perform an  $\alpha$ -transition into a process that satisfies  $\varphi$ .

[HM85] also contains the proof that this modal-logical characterisation of strong bisimilarity coincides with a characterisation that is effectively the same as the one we saw in [section 2.2](#) using strong bisimulations. Although they use different terminology, their result can be summarised as follows: for image-finite LTSs, two processes are strongly bisimilar iff they satisfy

the same set of HML formulas. We call this the *modal characterisation* of strong bisimilarity.

Let the *cardinality of conjunction* be the maximally allowed cardinality of sets of (sub-)formulas conjoined within a formula (for a given variant of HML). For the simple variant above, conjunction has finite cardinality. By allowing for conjunction of arbitrary cardinality (infinitary HML), the modal characterisation of strong bisimilarity can be proved for arbitrary LTSs. This is done in [appendix B](#).

In this section, however, conjunction is constrained to be of countable cardinality, as this turned out to be significantly easier to deal with in the upcoming proofs. The modal characterisation of strong bisimilarity, then, works for LTSs that are image-countable, as we shall see below.

Formulas  $\varphi$  are given by the following grammar, where  $I$  ranges over all subsets of the natural numbers:

$$\varphi ::= \bigwedge_{i \in I} \varphi_i \mid \neg \varphi \mid \langle \alpha \rangle \varphi$$

The semantics of HML formulas on LTSs are as above, with the alteration that a process satisfies  $\bigwedge_{i \in I} \varphi_i$  iff it satisfies  $\varphi_i$  for all  $i \in I$ .

Additional logical constants can be added as ‘syntactic sugar’:

$$\begin{aligned} tt &\equiv \bigwedge_{i \in \emptyset} \varphi_i \\ ff &\equiv \neg tt \\ \bigvee_{i \in I} \varphi_i &\equiv \neg \bigwedge_{i \in I} \neg \varphi_i \end{aligned}$$

---

Isabelle

---

## Syntax

By definition of countability, all countable sets of formulas can be given by  $\{\varphi_i\}_{i \in I} =: \Phi$  for some  $I \subseteq \mathbb{N}$  (then  $\bigwedge_{i \in I} \varphi_i$  shall correspond to  $\bigwedge \Phi$ ). Therefore, the following data type (parameterised by the type of actions ‘a’) formalises the definition of HML formulas above (`cset` is the type constructor for countable sets; `acset` and `rcset` are the type morphisms between the types `set` and `cset`; more details below).

I abstained from assigning the constructors a more readable symbolic notation because of the ambiguities and name clashes that would ensue in upcoming sections. The symbolic notations after the constructors below are just code comments.

```

datatype ('a)HML_formula =
  HML_conj  ⟨('a)HML_formula cset⟩ —  $\bigwedge \Phi$ 
| HML_neg   ⟨('a)HML_formula⟩ —  $\neg \varphi$ 
| HML_poss  ⟨'a⟩ ⟨('a)HML_formula⟩ —  $\langle \alpha \rangle \varphi$ 

```

The following abbreviations introduce useful constants as syntactic sugar (where `cimage HML_neg  $\Phi$`  corresponds to  $\{\neg \varphi \mid \varphi \in \Phi\}$ ).

```

abbreviation HML_true :: ⟨('a)HML_formula⟩ — #
  where ⟨HML_true  $\equiv$  HML_conj (acset  $\emptyset$ )⟩
abbreviation HML_false :: ⟨('a)HML_formula⟩ — ff
  where ⟨HML_false  $\equiv$  HML_neg HML_true⟩
abbreviation HML_disj :: ⟨('a)HML_formula cset  $\Rightarrow$  ('a)HML_formula⟩ —  $\bigvee \Phi$ 
  where ⟨HML_disj  $\Phi \equiv$  HML_neg (HML_conj (cimage HML_neg  $\Phi$ ))⟩

```

### Aside: The Type of Countable Sets

Since sets `set` and countable sets `cset` are different types in Isabelle, they have different membership relation terms. We introduce the following notation for membership of countable sets.

```

notation cin (⟨_  $\in_c$  _⟩ [100, 100] 100)

```

The following propositions should clarify how the type constructor `cset` and its morphisms are used. Note how the first proposition requires the assumption `countable X`, whereas the second one does not.

```

proposition
  fixes X :: ⟨'x set⟩
  assumes ⟨countable X⟩
  shows ⟨x  $\in$  X  $\iff$  x  $\in_c$  acset X⟩
  ⟨proof⟩
proposition
  fixes X :: ⟨'x cset⟩
  shows ⟨x  $\in_c$  X  $\iff$  x  $\in$  rcset X⟩
  ⟨proof⟩

```

### Semantics

The semantic satisfaction relation is formalised by the following function. Since the relation is not monotonic (due to negation terms), it cannot be directly defined in Isabelle as an inductive predicate, so we use the `function` command instead. This, then, requires us to prove that the function is well-defined (i.e. the function definition completely and compatibly covers all constructors of our data type) and total (i.e. it always terminates). It is easy to see that the former is the case for the function below.

**context** lts begin

```

function HML_sat :: ('s  $\Rightarrow$  ('a)HML_formula  $\Rightarrow$  bool)
  ( $\_ \models \_$ ) [50, 50] 50)
  where
    HML_sat_conj: ( $p \models \text{HML\_conj } \Phi$ ) = ( $\forall \varphi. \varphi \in_c \Phi \longrightarrow p \models \varphi$ )
  | HML_sat_neg: ( $p \models \text{HML\_neg } \varphi$ ) = ( $\neg p \models \varphi$ )
  | HML_sat_poss: ( $p \models \text{HML\_poss } \alpha \varphi$ ) = ( $\exists p'. p \mapsto_\alpha p' \wedge p' \models \varphi$ )
  <proof>

```

In order to prove that the function always terminates, we need to show that each sequence of recursive invocations reaches a base case<sup>2</sup> after finitely many steps. We do this by proving that the relation between process-formula pairs given by the recursive definition of the function is (contained within) a well-founded relation. A relation  $R \subseteq X \times X$  is called well-founded if each non-empty subset  $X' \subseteq X$  has a minimal element  $m$  that is not ' $R$ -greater' than any element of  $X'$ , i.e.  $\forall x \in X'. (x, m) \notin R$ . A property of well-founded relations is that all descending chains  $(x_0, x_1, x_2, \dots)$  (with  $(x_i, x_{i+1}) \in R$ ) starting at any element  $x_0 \in X$  are finite. This, then, implies that each sequence of recursive invocations terminates after finitely many steps.

These proofs were inspired by the Isabelle formalisations done in [WEP<sup>+</sup>16].

```

inductive_set HML_wf_rel :: (('s  $\times$  ('a)HML_formula) rel)
  where
    ( $\varphi \in_c \Phi \implies ((p, \varphi), (p, \text{HML\_conj } \Phi)) \in \text{HML\_wf\_rel}$ )
  | ( $((p, \varphi), (p, \text{HML\_neg } \varphi)) \in \text{HML\_wf\_rel}$ )
  | ( $((p, \varphi), (p', \text{HML\_poss } \alpha \varphi)) \in \text{HML\_wf\_rel}$ )

```

```

lemma HML_wf_rel_is_wf: (wf HML_wf_rel)
  <proof>

```

```

termination HML_sat using HML_wf_rel_is_wf
  by (standard, (simp add: HML_wf_rel.intros)+)

```

The semantic clauses for our additional constants are now easily derivable.

```

lemma HML_sat_true:
  shows ( $p \models \text{HML\_true}$ ) = True
  <proof>
lemma HML_sat_false:
  shows ( $p \models \text{HML\_false}$ ) = False
  <proof>
lemma HML_sat_disj:
  shows ( $p \models \text{HML\_disj } \Phi$ ) = ( $\exists \varphi. \varphi \in_c \Phi \wedge p \models \varphi$ )
  <proof>

```

---

<sup>2</sup>For our satisfaction function, the recursive base case is, of course, the empty conjunction, since  $\forall \varphi. \varphi \in \emptyset \longrightarrow p \models \varphi$  is a tautology.



### Modal Characterisation of Strong Bisimilarity

First, we introduce HML-equivalence as follows.

```
definition HML_equivalent :: ('s  $\Rightarrow$  's  $\Rightarrow$  bool)
  where (HML_equivalent p q  $\equiv$  ( $\forall \varphi. (p \models \varphi) \longleftrightarrow (q \models \varphi)$ ))
```

Since formulas are closed under negation, the following lemma holds.

```
lemma distinguishing_formula:
  assumes  $\neg$  HML_equivalent p q
  shows  $\exists \varphi. p \models \varphi \wedge \neg q \models \varphi$ 
  <proof>
```

HML-equivalence is clearly symmetrical.

```
lemma HML_equivalent_symm:
  assumes (HML_equivalent p q)
  shows (HML_equivalent q p)
  <proof>
```

We can now formally prove the modal characterisation of strong bisimilarity, i.e.: two processes are HML-equivalent iff they are strongly bisimilar. The proof is borrowed from [AILS07, theorem 5.1].

I chose to include these proofs in the thesis document, because they translate quite beautifully, in my opinion, and are comparatively short.

We show the  $\Rightarrow$ -case first, by induction over  $\varphi$ .

```
lemma strong_bisimilarity_implies_HM_equivalence:
  assumes  $p \leftrightarrow q$   $p \models \varphi$ 
  shows  $q \models \varphi$ 
  using assms
proof (induct  $\varphi$  arbitrary: p q)
  case (HML_conj  $\Phi$ )
  then show ?case
    using HML_sat_conj by (meson cin.rep_eq)
next
  case (HML_neg  $\varphi$ )
  then show ?case
    using HML_sat_neg strongly_bisimilar_symm by meson
next
  case (HML_poss  $\alpha \varphi$ )
  then show ?case
    using HML_sat_poss strongly_bisimilar_step(1) by meson
qed
```

Before we can show the  $\Leftarrow$ -case, we need to prove the following lemma: for some binary predicate  $P$ , if for every element  $a$  of a set  $A$ , there exists an element  $x$  such that  $P(a, x)$  is true, then we can obtain a set  $X$  that contains these  $x$  (for all  $a \in A$ ) and has the same cardinality as  $A$ .

Since more than one  $x$  might exist for each  $a$  such that  $P(a, x)$  is true, the set  $\{x \mid a \in A \wedge P(a, x)\}$  might have greater cardinality than  $A$ . In order to obtain a set  $X$  of same cardinality as  $A$ , we need to invoke the axiom of choice in our proof.

```

lemma obtaining_set:
  assumes
     $\langle \forall a \in A. \exists x. P\ a\ x \rangle$ 
     $\langle \text{countable } A \rangle$ 
  obtains  $X$  where
     $\langle \forall a \in A. \exists x \in X. P\ a\ x \rangle$ 
     $\langle \forall x \in X. \exists a \in A. P\ a\ x \rangle$ 
     $\langle \text{countable } X \rangle$ 
proof
  — the SOME operator (Hilbert's selection operator  $\varepsilon$ ) invokes the axiom of choice
  define  $xm$  where  $\langle xm \equiv \lambda a. \text{SOME } x. P\ a\ x \rangle$ 
  define  $X$  where  $\langle X \equiv \{xm\ a \mid a. a \in A\} \rangle$ 

  show  $\langle \forall a \in A. \exists x \in X. P\ a\ x \rangle$ 
    using  $X\_def\ xm\_def\ assms(1)$   $\langle proof \rangle$ 
  show  $\langle \forall x \in X. \exists a \in A. P\ a\ x \rangle$ 
    using  $X\_def\ xm\_def\ assms(1)$   $\langle proof \rangle$ 
  show  $\langle \text{countable } X \rangle$ 
    using  $X\_def\ xm\_def\ assms(2)$   $\langle proof \rangle$ 
qed

```

We can now show, assuming image-countability of the given LTS, that HML-equivalence is a strong bisimulation. The proof utilises classical contradiction: if HML-equivalence were no strong bisimulation, there would be some processes  $p$  and  $q$  that are HML-equivalent, with  $p \xrightarrow{\alpha} p'$  for some  $p'$  (i.e.  $p' \in \text{Der}(p, \alpha)$ ), but for all  $q' \in \text{Der}(q, \alpha)$ ,  $p'$  and  $q'$  are not HML-equivalent. Then, for each  $q' \in \text{Der}(q, \alpha)$ , there would be a distinguishing formula  $\varphi_{q'}$  which  $p'$  satisfies but  $q'$  does not. Using our `obtaining_set` lemma, we can obtain the set  $\Phi = \{\varphi_{q'}\}_{q' \in \text{Der}(q, \alpha)}$ , which is countable, since  $\text{Der}(q, \alpha)$  is countable, by the image-countability assumption. Since we allow for conjunction of countable cardinality,  $\bigwedge \Phi$  is a valid formula. By construction,  $p$  can make an  $\alpha$ -transition into a state that satisfies  $\bigwedge \Phi$  (i.e.  $p \models \langle \alpha \rangle \bigwedge \Phi$ ), whereas  $q$  cannot (i.e.  $q \not\models \langle \alpha \rangle \bigwedge \Phi$ ). This is a contradiction, since, by assumption,  $p$  and  $q$  are HML-equivalent. Therefore, HML-equivalence must be a strong bisimulation.

```

lemma HML_equivalence_is_SB:
  assumes
    ⟨image_countable⟩
  shows
    ⟨SB HML_equivalent⟩
proof -
  {
    fix p q p' α
    assume ⟨HML_equivalent p q⟩
    assume ⟨p ⟶α p'⟩
    assume ⟨∀ q' ∈ Der(q, α). ¬ HML_equivalent p' q'⟩

    hence exists_φq': ⟨∀ q' ∈ Der(q, α). ∃ φ. p' ⊨ φ ∧ ¬ q' ⊨ φ⟩
      using distinguishing_formula by blast

    from ⟨image_countable⟩ have ⟨countable Der(q, α)⟩
      using image_countable_def by simp

    from obtaining_set[
      where ?A = ⟨Der(q, α)⟩
      and ?P = ⟨λ q' φ. p' ⊨ φ ∧ ¬ q' ⊨ φ⟩,
      OF exists_φq' ⟨countable Der(q, α)⟩]
    obtain Φ where *:
      ⟨∀ φ ∈ Φ. ∃ q' ∈ Der(q, α). p' ⊨ φ ∧ ¬ q' ⊨ φ⟩
      ⟨∀ q' ∈ Der(q, α). ∃ φ ∈ Φ. p' ⊨ φ ∧ ¬ q' ⊨ φ⟩
      ⟨countable Φ⟩
      by (this, blast+)

    have ⟨p ⊨ HML_poss α (HML_conj (acset Φ))⟩
      using ⟨p ⟶α p'⟩ *(1,3) HML_sat.simps(1,3)
      acset_inverse mem_Collect_eq cin.rep_eq
      by metis

    moreover have ⟨¬ q ⊨ HML_poss α (HML_conj (acset Φ))⟩
      using *(2,3) cin.rep_eq
      by fastforce

    ultimately have False
      using ⟨HML_equivalent p q⟩
      unfolding HML_equivalent_def
      by meson
  }

  — We showed the case for p ⟶α p', but not q ⟶α q'.
  — Clearly, this case is covered by the symmetry of HML-equivalence.
  from this show ⟨SB HML_equivalent⟩ unfolding SB_def
    using HML_equivalent_symm by blast
qed

```

We can now conclude the modal characterisation of strong bisimilarity.

```

theorem modal_characterisation_of_strong_bisimilarity:
  assumes ⟨image_countable⟩
  shows ⟨(p ↔ q) ⟷ (∀ φ. p ⊨ φ ⟷ q ⊨ φ)⟩
proof
  show ⟨(p ↔ q) ⟹ ∀φ. (p ⊨ φ) = (q ⊨ φ)⟩
    using strong_bisimilarity_implies_HM_equivalence
    strongly_bisimilar_symm
    by blast
  next
  show ⟨∀φ. (p ⊨ φ) = (q ⊨ φ) ⟹ (p ↔ q)⟩
    using HML_equivalence_is_SB[OF assms]
    HML_equivalent_def strongly_bisimilar_def
    by blast
qed

end — of context lts

```


## 2.4 Labelled Transition Systems with Time-Outs

---

In addition to the hidden action  $\tau$ , labelled transition systems with time-outs ( $\text{LTS}_t$ ) [vG21] include the *time-out action*  $t$  as another special action;  $t$ -transitions can only be performed when no other (non-time-out) transition is allowed in a given environment. The rationale is that, in this model, all transitions that are facilitated by or independent of the environment happen instantaneously, and only when no such transition is possible, time elapses and the system is idle. However, since the passage of time is not quantified here, the system does not *have* to take a time-out transition in such a case; instead, the environment can spontaneously change its set of allowed actions (corresponding to a time-out on the part of the environment). Thus, the resolution of an idling period is non-deterministic.


In most works on LTSs, the actions which the environment allows in any given moment are usually not modelled explicitly; an (often implicit) requirement for any property of the system is that it should hold for arbitrary (and arbitrarily changing) environments. The introduction of time-outs necessitates an explicit consideration of the environment, as the possibility of a transition not only depends on whether its labelling action is currently allowed, but potentially on the set of all actions currently allowed by the environment. This is why, in previous sections, I have put unusual emphasis on the actions that are allowed or blocked by the system's environment in a given moment. Henceforth, I will refer to 'environments allowing *exactly* the actions in  $X$ ' simply as 'environments  $X$ '.

**Example** The process  $p$  can perform an  $a$ -transition in environments allowing the action  $a$  and a  $t$ -transition in environments blocking  $a$ . On the other hand, the process  $q$  cannot perform a  $t$ -transition in any environment, since the  $\tau$ -transition will always be performed immediately.




Furthermore, since  $t$ -transitions (as well as  $\tau$ -transitions) are hidden, they cannot trigger a change of the environment, so some states may only ever be entered in certain environments.

**Example** The process  $p$  can perform a  $t$ -transition only in environments blocking  $a$ . Therefore, the subsequent state  $p_2$  must be entered in such an environment. The  $\tau$ -transition is now the only possible transition and will always be performed immediately.



On the other hand, transitions with labels other than  $\tau$  and  $t$  require an interaction with the environment, and therefore *are* detectable and can trigger a change in the allowed actions of the environment.


**Example** Only in environments blocking  $a$ ,  $p$  can make a  $t$ -transition to  $p_2$ . However (if  $b$  is allowed), the performance of the  $b$ -transition into  $p_3$  may trigger a change of the environment, so it is possible that  $p_3$  could perform its  $a$ -transition.




These semantics of  $\text{LTS}_t$ s induce a novel notion of behavioural equivalence, which will be discussed in the next section.

### Practical Example

As in [section 2.1](#), we shall consider a real-world machine that may be modelled as an  $LTS_t$ . Continuing with our example, let us imagine that our simple vending machine ejects the coin if no snack has been selected after a certain amount of time. We can attempt to model the machine with this extended behaviour as an LTS, where the coin ejection requires no interaction and is therefore also modelled as a  $\tau$ -transition.



However, this LTS also models a machine that randomly ejects coins right after insertion. In order to distinguish these behaviours, we need a  $t$ -transition along with  $LTS_t$  semantics.




---

Isabelle

---

We extend LTSs with hidden actions (`lts_tau`) by the special action  $\mathfrak{t}$ . We have to explicitly require (/assume) that  $\tau \neq \mathfrak{t}$ ; when instantiating the locale `lts_timeout` and specifying a concrete type for the type variable 'a, this assumption must be (proved to be) satisfied.

```

locale lts_timeout = lts_tau tran  $\tau$ 
  for tran :: 's  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool
    ( $\_ \mapsto \_$  [70, 70, 70] 80)
    and  $\tau$  :: 'a +
  fixes t :: 'a
  assumes tau_not_t:  $\langle \tau \neq t \rangle$ 
begin

```

We define the set of (relevant) visible actions (denoted by  $A \subseteq Act$ ) as the set of all actions that are not hidden and that are labels of some transition in the given LTS.

```

definition visible_actions :: <'a set>
  where <visible_actions
     $\equiv \{a. (a \neq \tau) \wedge (a \neq t) \wedge (\exists p p'. p \mapsto_a p')\}$ >

```

The formalisations in upcoming sections will often involve the type 'a set option, which has values of the form None and Some X for some  $X :: 'a \text{ set}$ . We will use the metavariable XoN (for 'X or None'). The following abbreviation will be useful in these situations.

```

abbreviation some_visible_subset :: <'a set option  $\Rightarrow$  bool>
  where <some_visible_subset XoN
     $\equiv \exists X. \text{XoN} = \text{Some } X \wedge X \subseteq \text{visible\_actions}$ >

```

The initial actions of a process ( $\mathcal{I}(p)$  in [vG20]) are the actions for which the process has a transition it can perform immediately (if facilitated by the environment), i.e. it is not a  $t$ -transition.

```

definition initial_actions :: <'s  $\Rightarrow$  'a set>
  where <initial_actions(p)
     $\equiv \{\alpha. (\alpha \in \text{visible\_actions} \vee \alpha = \tau) \wedge (\exists p'. p \mapsto_\alpha p')\}$ >

```

In [vG20], the term  $\mathcal{I}(p) \cap (X \cup \{\tau\}) = \emptyset$  is used a lot, which expresses that there are no immediate transitions the process  $p$  can perform (i.e. it is idle) in environments  $X$ .

```

abbreviation idle :: <'s  $\Rightarrow$  'a set  $\Rightarrow$  bool>
  where <idle p X  $\equiv \text{initial\_actions}(p) \cap (X \cup \{\tau\}) = \emptyset$ >

```

The following corollary is an immediate consequence of this definition.

**corollary** `idle_no_derivatives:`

**assumes**

$\langle \text{idle } p \ X \rangle$

$\langle X \subseteq \text{visible\_actions} \rangle$

$\langle \alpha \in X \cup \{\tau\} \rangle$

**shows**

$\langle \# \ p'. \ p \mapsto \alpha \ p' \rangle$


*(proof)*

**end** — of locale `lts_timeout`

## 2.5 Reactive Bisimilarity

In the examples of the previous section, we saw that there are  $\text{LTS}_t$ s with transitions that can never be performed or that can only be performed in certain environments. The behavioural equivalence implied hereby is defined in [vG20] as *strong reactive bisimilarity*.

**Example** The processes  $p$  and  $q$  are behaviourally equivalent for  $\text{LTS}_t$  semantics, i.e. strongly reactive bisimilar.



### Strong Reactive Bisimulations

Van Glabbeek introduces several characterisations of this equivalence, beginning with *strong reactive bisimulation* (SRB) relations. These differ from strong bisimulations in that the relations contain not only pairs of processes,  $(p, q)$ , but additionally triples consisting of two processes and a set of actions,  $(p, X, q)$ . The following definition of SRBs is quoted, with minor adaptations, from [vG20, definition 1]:

A *strong reactive bisimulation* is a symmetric relation

$$\mathcal{R} \subseteq (\text{Proc} \times \mathcal{P}(A) \times \text{Proc}) \cup (\text{Proc} \times \text{Proc})$$

(meaning that  $(p, X, q) \in \mathcal{R} \iff (q, X, p) \in \mathcal{R}$  and  $(p, q) \in \mathcal{R} \iff (q, p) \in \mathcal{R}$ ), such that,



for all  $(p, q) \in \mathcal{R}$ :

1. if  $p \xrightarrow{\tau} p'$ , then  $\exists q'$  such that  $q \xrightarrow{\tau} q'$  and  $(p', q') \in \mathcal{R}$ ,
2.  $(p, X, q) \in \mathcal{R}$  for all  $X \subseteq A$ ,

and for all  $(p, X, q) \in \mathcal{R}$ :

3. if  $p \xrightarrow{a} p'$  with  $a \in X$ , then  $\exists q'$  such that  $q \xrightarrow{a} q'$  and  $(p', q') \in \mathcal{R}$ ,
4. if  $p \xrightarrow{\tau} p'$ , then  $\exists q'$  such that  $q \xrightarrow{\tau} q'$  and  $(p', X, q') \in \mathcal{R}$ ,
5. if  $\mathcal{I}(p) \cap (X \cup \{\tau\}) = \emptyset$ , then  $(p, q) \in \mathcal{R}$ , and
6. if  $\mathcal{I}(p) \cap (X \cup \{\tau\}) = \emptyset$  and  $p \xrightarrow{t} p'$ , then  $\exists q'$  such that  $q \xrightarrow{t} q'$  and  $(p', X, q') \in \mathcal{R}$ .

We can derive the following intuitions: an environment can either be stable, allowing a specific set of actions, or indeterminate. Indeterminate environments cannot facilitate any transitions, but they can stabilise into arbitrary stable environments. This is expressed by clause 2. Hence,  $X$ -bisimilarity is behavioural equivalence in stable environments  $X$ , and reactive bisimilarity is behavioural equivalence in indeterminate environments (and thus in arbitrary stable environments).

Since only stable environments can facilitate transitions, there are no clauses involving visible action transitions for  $(p, q) \in \mathcal{R}$ . However,  $\tau$ -transitions can be performed regardless of the environment, hence clause 1.

At this point, it is important to discuss what exactly it means for an action to be visible or hidden in this context: as we saw in the last section, the environment cannot react (change its set of allowed actions) when the system performs a  $\tau$ - or a  $t$ -transition, since these are hidden actions. However, since we are talking about a *strong* bisimilarity (as opposed to e.g. weak bisimilarity), the performance of  $\tau$ - or  $t$ -transitions is still relevant when examining and comparing the behavior of systems.

With that, we can look more closely at the remaining clauses: in clause 3, given  $(p, X, q) \in \mathcal{R}$ , for  $p \xrightarrow{a} p'$  with  $a \in X$ , we require for the ‘mirroring’ state  $q'$  that  $(p', q') \in \mathcal{R}$ , because  $a$  is a visible action and the transition can thus trigger a change of the environment;<sup>3</sup> on the other hand, in clause 4, for  $p \xrightarrow{\tau} p'$ , and in clause 6, for  $p \xrightarrow{t} p'$ , we require  $(p', X, q') \in \mathcal{R}$ , because these actions are hidden and cannot trigger a change of the environment.

Lastly, clause 5 formalises the possibility of the environment timing out (i.e. turning into an indeterminate environment) instead of the system.

These intuitions also form the basis for the process mapping which will be presented in [section 3.1](#).

---

<sup>3</sup>This is why van Glabbeek talks about *triggered* environments rather than indeterminate ones. I will use both terms interchangeably.

### Strong Reactive/ $X$ -Bisimilarity

Two processes  $p$  and  $q$  are *strongly reactive bisimilar* ( $p \leftrightarrow_r q$ ) iff there is an SRB containing  $(p, q)$ , and *strongly  $X$ -bisimilar* ( $p \leftrightarrow_r^X q$ ), i.e. equivalent in environments  $X$ , when there is an SRB containing  $(p, X, q)$ .

### Generalised Strong Reactive Bisimulations

Another characterisation of reactive bisimilarity uses *generalised strong reactive bisimulation* (GSRB) relations [vG20, definition 3], defined over the same set as SRBs. It is proved that both characterisations do, in fact, characterise the same equivalence. More details will be discussed in the formalisation below.

---

Isabelle

---

We first formalise both SRB and GSRB relations (as well as strong reactive bisimilarity, defined by the existence of an SRB, as above), and then replicate the proof of their correspondence.

### Strong Reactive Bisimulations

SRB relations are defined over the set

$$(Proc \times \mathcal{P}(A) \times Proc) \cup (Proc \times Proc).$$

As can be easily seen, this set is isomorphic to

$$(Proc \times (\mathcal{P}(A) \cup \{\perp\}) \times Proc),$$

which is a subset of

$$(Proc \times (\mathcal{P}(Act) \cup \{\perp\}) \times Proc).$$

This last set can now be easily formalised in terms of a type, where we formalise  $\mathcal{P}(Act) \cup \{\perp\}$  as `'a set option`.

The fact that SRBs are defined using the power set of visible actions ( $A$ ), whereas our type uses all actions ( $Act$  / `'a`), is handled by the first line of the definition below. The second line formalises that symmetry is required by definition. All other lines are direct formalisations of the clauses of the original definition.

**context** lts\_timeout **begin**

— strong reactive bisimulation [vG20, definition 1]

**definition** SRB ::  $\langle 's \Rightarrow 'a \text{ set option} \Rightarrow 's \Rightarrow \text{bool} \rangle \Rightarrow \text{bool}$   
**where**  $\langle \text{SRB } R \equiv$   
 $(\forall p \ X \ q. \ R \ p \ (\text{Some } X) \ q \longrightarrow X \subseteq \text{visible\_actions}) \wedge$   
 $(\forall p \ XoN \ q. \ R \ p \ XoN \ q \longrightarrow R \ q \ XoN \ p) \wedge$   
 $(\forall p \ q. \ R \ p \ \text{None } q \longrightarrow$   
 $(\forall p'. \ p \mapsto_{\tau} p' \longrightarrow (\exists q'. (q \mapsto_{\tau} q') \wedge R \ p' \ \text{None } q')) \wedge$   
 $(\forall X \subseteq \text{visible\_actions}. (R \ p \ (\text{Some } X) \ q))) \wedge$   
 $(\forall p \ X \ q. \ R \ p \ (\text{Some } X) \ q \longrightarrow$   
 $(\forall p' \ a. \ p \mapsto_a p' \wedge a \in X \longrightarrow (\exists q'. (q \mapsto_a q') \wedge$   
 $R \ p' \ \text{None } q')) \wedge$   
 $(\forall p'. \ p \mapsto_{\tau} p' \longrightarrow (\exists q'. (q \mapsto_{\tau} q') \wedge R \ p' \ (\text{Some } X) \ q')) \wedge$   
 $(\text{idle } p \ X \longrightarrow R \ p \ \text{None } q) \wedge$   
 $(\forall p'. \ \text{idle } p \ X \wedge (p \mapsto_t p') \longrightarrow (\exists q'. \ q \mapsto_t q' \wedge$   
 $R \ p' \ (\text{Some } X) \ q')) \rangle$

### Strong Reactive/ $X$ -Bisimilarity

Van Glabbeek differentiates between strong reactive bisimilarity  $((p, q) \in \mathcal{R}$  for an SRB  $\mathcal{R}$ ) and strong  $X$ -bisimilarity  $((p, X, q) \in \mathcal{R}$  for an SRB  $\mathcal{R}$ ).

**definition** strongly\_reactive\_bisimilar ::  $\langle 's \Rightarrow 's \Rightarrow \text{bool} \rangle$   
 $\langle \_ \leftrightarrow_r \_ \rangle [70, 70] \ 70)$   
**where**  $\langle p \leftrightarrow_r q \equiv \exists R. \text{SRB } R \wedge R \ p \ \text{None } q \rangle$

**definition** strongly\_X\_bisimilar ::  $\langle 's \Rightarrow 'a \text{ set} \Rightarrow 's \Rightarrow \text{bool} \rangle$   
 $\langle \_ \leftrightarrow_r^X \_ \rangle [70, 70, 70] \ 70)$   
**where**  $\langle p \leftrightarrow_r^X q \equiv \exists R. \text{SRB } R \wedge R \ p \ (\text{Some } X) \ q \rangle$

For the upcoming proofs, it is useful to combine both reactive and  $X$ -bisimilarity into a single relation.

**definition** strongly\_reactive\_or\_X\_bisimilar  
::  $\langle 's \Rightarrow 'a \text{ set option} \Rightarrow 's \Rightarrow \text{bool} \rangle$   
**where**  $\langle \text{strongly\_reactive\_or\_X\_bisimilar } p \ XoN \ q$   
 $\equiv \exists R. \text{SRB } R \wedge R \ p \ XoN \ q \rangle$

Obviously, then, these relations coincide accordingly.

**corollary**  $\langle p \leftrightarrow_r q \iff \text{strongly\_reactive\_or\_X\_bisimilar } p \ \text{None } q \rangle$   
 $\langle \text{proof} \rangle$

**corollary**  $\langle p \leftrightarrow_r^X q \iff \text{strongly\_reactive\_or\_X\_bisimilar } p \ (\text{Some } X) \ q \rangle$   
 $\langle \text{proof} \rangle$

### Generalised Strong Reactive Bisimulations

Since GSRBs are defined over the same set as SRBs, the same considerations concerning the type and the clauses of the definition as above hold.

— generalised strong reactive bisimulation [vG20, definition 3]

**definition**  $\text{GSRB} :: \langle ('s \Rightarrow 'a \text{ set option} \Rightarrow 's \Rightarrow \text{bool}) \Rightarrow \text{bool} \rangle$   
**where**  $\langle \text{GSRB } R \equiv$   
 $(\forall p \ X \ q. \ R \ p \ (\text{Some } X) \ q \longrightarrow X \subseteq \text{visible\_actions}) \wedge$   
 $(\forall p \ X \text{on } q. \ R \ p \ X \text{on } q \longrightarrow R \ q \ X \text{on } p) \wedge$   
 $(\forall p \ q. \ R \ p \ \text{None } q \longrightarrow$   
 $(\forall p' \ a. \ p \mapsto_a p' \wedge a \in \text{visible\_actions} \cup \{\tau\} \longrightarrow$   
 $(\exists q'. \ q \mapsto_a q' \wedge R \ p' \ \text{None } q')) \wedge$   
 $(\forall X \ p'. \ \text{idle } p \ X \wedge X \subseteq \text{visible\_actions} \wedge p \mapsto_t p' \longrightarrow$   
 $(\exists q'. \ q \mapsto_t q' \wedge R \ p' \ (\text{Some } X) \ q')) \rangle \wedge$   
 $(\forall p \ Y \ q. \ R \ p \ (\text{Some } Y) \ q \longrightarrow$   
 $(\forall p' \ a. \ a \in \text{visible\_actions} \wedge p \mapsto_a p' \wedge (a \in Y \vee \text{idle } p \ Y) \longrightarrow$   
 $(\exists q'. \ q \mapsto_a q' \wedge R \ p' \ \text{None } q')) \wedge$   
 $(\forall p'. \ p \mapsto_\tau p' \longrightarrow$   
 $(\exists q'. \ q \mapsto_\tau q' \wedge R \ p' \ (\text{Some } Y) \ q')) \wedge$   
 $(\forall p' \ X. \ \text{idle } p \ (X \cup Y) \wedge X \subseteq \text{visible\_actions} \wedge p \mapsto_t p' \longrightarrow$   
 $(\exists q'. \ q \mapsto_t q' \wedge R \ p' \ (\text{Some } X) \ q')) \rangle \rangle$

### GSRBs characterise strong reactive/ $X$ -bisimilarity

[vG20, proposition 4] reads (notation adapted): ‘ $p \leftrightarrow_r q$  iff there exists a GSRB  $\mathcal{R}$  with  $(p, q) \in \mathcal{R}$ . Likewise,  $p \leftrightarrow_r^X q$  iff there exists a GSRB  $\mathcal{R}$  with  $(p, X, q) \in \mathcal{R}$ .’ We shall now replicate the proof of this proposition. First, we prove that each SRB is a GSRB (by showing that each SRB satisfies all clauses of the definition of GSRBs).

**lemma**  $\text{SRB\_is\_GSRB}$ :

**assumes**  $\langle \text{SRB } R \rangle$

**shows**  $\langle \text{GSRB } R \rangle$

$\langle \text{proof} \rangle$

Then, we show that each GSRB can be extended to yield an SRB. First, we define this extension. Generally, GSRBs can be smaller than SRBs when proving reactive bisimilarity of processes, because they require triples  $(p, X, q)$  only after encountering  $t$ -transitions, whereas SRBs require these triples for all processes and all environments. Furthermore, some process pairs  $(p, q)$  related to environment time-outs are also omitted in GSRBs. These tuples are re-added by this extension.

```

definition GSRB_extension
  :: ⟨('s⇒'a set option⇒'s ⇒ bool)⇒('s⇒'a set option⇒'s ⇒ bool)⟩
  where ⟨(GSRB_extension R) p XoN q ≡
    (R p XoN q)
    ∨ (some_visible_subset XoN ∧ R p None q)
    ∨ ((XoN = None ∨ some_visible_subset XoN)
      ∧ (∃ Y. R p (Some Y) q ∧ idle p Y))⟩

```

Now we show that this extension does, in fact, yield an SRB (again, by showing that all clauses of the definition of SRBs are satisfied).

```

lemma GSRB_extension_is_SRB:
  assumes
    ⟨GSRB R⟩
  shows
    ⟨SRB (GSRB_extension R)⟩ (is ⟨SRB ?R_ext⟩)
  ⟨proof⟩

```

Finally, we can conclude the following:

```

lemma GSRB_whenever_SRB:
  shows ⟨(∃ R. GSRB R ∧ R p XoN q) ⟷ (∃ R. SRB R ∧ R p XoN q)⟩
  ⟨proof⟩

```

This, now, directly implies that GSRBs do characterise strong reactive/ $X$ -bisimilarity.

```

proposition GSRBs_characterise_strong_reactive_bisimilarity:
  ⟨p ↔r q ⟷ (∃ R. GSRB R ∧ R p None q)⟩
  using GSRB_whenever_SRB strongly_reactive_bisimilar_def by blast

```

```

proposition GSRBs_characterise_strong_X_bisimilarity:
  ⟨p ↔rX q ⟷ (∃ R. GSRB R ∧ R p (Some X) q)⟩
  using GSRB_whenever_SRB strongly_X_bisimilar_def by blast

```

```

end — of context lts_timeout

```

As a little meta-comment, I would like to point out that van Glabbeek's proof spans a total of five lines ('Clearly, [...]', 'It is straightforward to check [...]', whereas the Isabelle proof takes up around 250 lines of code. This just goes to show that for things which are clear and straightforward for humans, it might require quite some effort to 'explain' them to a computer.

## 2.6 Hennessy-Milner Logic with Time-Outs

---

In [vG20, section 3], van Glabbeek extends Hennessy-Milner logic by a family of new modal operators  $\langle X \rangle \varphi$  for  $X \subseteq A$ , as well as additional satisfaction relations  $\models_X$  for  $X \subseteq A$ . Intuitively,  $p \models \langle X \rangle \varphi$  means that  $p$  is idle when placed in an environment  $X$  and  $p$  can perform a  $t$ -transition into a state that satisfies  $\varphi$ ;  $p \models_X \varphi$  means that  $p$  satisfies  $\varphi$  in environments  $X$ .

I will refer to this extension as *Hennessy-Milner Logic with Time-Outs* ( $\text{HML}_t$ ) and to  $\langle X \rangle$  for  $X \subseteq A$  as the *time-out-possibility operators* (to be distinguished from the ordinary possibility operators  $\langle \alpha \rangle$  for  $\alpha \in \text{Act}$ ).

The precise semantics are given by the following inductive definition of the satisfaction relation [vG20, section 3] (notation adapted):

$$\begin{array}{ll}
p \models \bigwedge_{i \in I} \varphi_i & \text{if } \forall i \in I. p \models \varphi_i \\
p \models \neg \varphi & \text{if } p \not\models \varphi \\
p \models \langle \alpha \rangle \varphi \quad \text{with } \alpha \in A \cup \{\tau\} & \text{if } \exists p'. p \xrightarrow{\alpha} p' \wedge p' \models \varphi \\
p \models \langle X \rangle \varphi \quad \text{with } X \subseteq A & \text{if } \mathcal{I}(p) \cap (X \cup \{\tau\}) = \emptyset \wedge \exists p'. p \xrightarrow{t} p' \wedge p' \models_X \varphi \\
\\
p \models_X \bigwedge_{i \in I} \varphi_i & \text{if } \forall i \in I. p \models_X \varphi_i \\
p \models_X \neg \varphi & \text{if } p \not\models_X \varphi \\
p \models_X \langle a \rangle \varphi \quad \text{with } a \in A & \text{if } a \in X \wedge \exists p'. p \xrightarrow{a} p' \wedge p' \models \varphi \\
p \models_X \langle \tau \rangle \varphi & \text{if } \exists p'. p \xrightarrow{\tau} p' \wedge p' \models_X \varphi \\
p \models_X \varphi & \text{if } \mathcal{I}(p) \cap (X \cup \{\tau\}) = \emptyset \wedge p \models \varphi
\end{array}$$

The same intuitions regarding triggered and stable environments as for the definition of strong reactive bisimulations in section 2.2 hold.  $\models$  expresses that a property holds in indeterminate environments and  $\models_X$  that a property holds in stable environments  $X$ . The last clause expresses the possibility of stable environments timing out into triggered environments.

Van Glabbeek then also proves that  $\text{HML}_t$  characterises strong reactive/ $X$ -bisimilarity, i.e. that  $p \leftrightarrow_r q \iff (\forall \varphi. p \models \varphi \iff q \models \varphi)$  and  $p \leftrightarrow_r^X q \iff (\forall \varphi. p \models_X \varphi \iff q \models_X \varphi)$ , where  $\varphi$  are formulas of  $\text{HML}_t$ . A replication of the proof of this characterisation, however, is not part of this thesis.

---

Isabelle

---

The following formalisation is analogous to the one in section 2.3.

```

datatype ('a)HMLt_formula =
  HMLt_conj <('a)HMLt_formula cset> —  $\bigwedge \Phi$ 
| HMLt_neg <('a)HMLt_formula> —  $\neg \varphi$ 
| HMLt_poss <'a> <('a)HMLt_formula> —  $\langle \alpha \rangle \varphi$ 
| HMLt_time <'a set> <('a)HMLt_formula> —  $\langle X \rangle \varphi$ 

```

In order to formalise the semantics, I combined both the usual satisfaction relation  $\models$  and the environment satisfaction relations  $\models_X$  into one predicate, which is formalised by the function `HMLt_sat` below, where  $p \models_{\text{None}} \varphi$  corresponds to  $p \models \varphi$  and  $p \models_{\text{Some } X} \varphi$  corresponds to  $p \models_X \varphi$ .

Note that, in Isabelle code, I use the symbol  $\models$  for all satisfaction relations in the context of  $\text{HML}_t$ , whereas I use  $\models$  for satisfaction relations in the context of ordinary HML. This notational nuance will be important when we examine the relationship between the satisfaction relations of  $\text{HML}_t$  and HML in the context of the reduction in [section 3.4](#).

The first four clauses of my formalisation are clearly direct translations of the clauses for the satisfaction relation  $\models$  above. It is less easy to see that the next four clauses do, in fact, correspond to the five clauses for  $\models_X$ .

First, each of the four clauses below includes the requirement that  $X$  is a subset of the visible actions; in the original definition, the satisfaction relations  $\models_X$  are only defined for those  $X$  to begin with.

Next, the clause for  $p \models_{\text{Some } X} (\text{HMLt\_poss } \alpha \varphi)$  combines the original clauses for  $p \models_X \langle a \rangle \varphi$  and  $p \models_X \langle \tau \rangle \varphi$ .

Lastly and most importantly, the last clause of the original definition, stating that  $p \models_X \varphi$  if  $p$  is idle in environments  $X$  and  $p \models \varphi$ , is added disjunctively to the cases  $p \models_{\text{Some } X} (\text{HMLt\_poss } \alpha \varphi)$  and  $p \models_{\text{Some } X} (\text{HMLt\_time } Y \varphi)$ ; the latter case is not part of the original definition and can only be true by virtue of the last clause of the original definition, wherefore this is the only way for this case in the function definition below to be true.

I will show below that this is sufficient to assure that my satisfaction function satisfies the last clause of the original definition, i.e. that it is not required to be added disjunctively to the cases  $p \models_{\text{Some } X} (\text{HMLt\_conj } \Phi)$  and  $p \models_{\text{Some } X} (\text{HMLt\_neg } \varphi)$ .

**context** `lts_timeout` **begin**

```
function HMLt_sat :: ('s=>'a set option=>('a)HMLt_formula => bool)
  (<_ <=>[_] _> [50, 50, 50] 50)
  where
    <(p <=>[None] (HMLt_conj Φ))> =
      <(∀ φ. φ ∈c Φ → p <=>[None] φ)>
  | <(p <=>[None] (HMLt_neg φ))> =
      <(¬ p <=>[None] φ)>
  | <(p <=>[None] (HMLt_poss α φ))> =
      <((α ∈ visible_actions ∪ {τ}) ∧
        (∃ p'. p →α p' ∧ p' <=>[None] φ))>
  | <(p <=>[None] (HMLt_time X φ))> =
      <((X ⊆ visible_actions) ∧ (idle p X) ∧
        (∃ p'. p →t p' ∧ p' <=>[Some X] φ))>
```

```

| ⟨(p  $\models$ ?[Some X] (HMLt_conj  $\Phi$ )) = (X  $\subseteq$  visible_actions  $\wedge$ 
  ( $\forall \varphi. \varphi \in_c \Phi \longrightarrow p \models$ ?[Some X]  $\varphi$ ))⟩
| ⟨(p  $\models$ ?[Some X] (HMLt_neg  $\varphi$ )) = (X  $\subseteq$  visible_actions  $\wedge$ 
  ( $\neg p \models$ ?[Some X]  $\varphi$ ))⟩
| ⟨(p  $\models$ ?[Some X] (HMLt_poss  $\alpha \varphi$ )) = (X  $\subseteq$  visible_actions  $\wedge$ 
  ((( $\alpha \in X$ )  $\wedge$  ( $\exists p'. p \mapsto_\alpha p' \wedge p' \models$ ?[None]  $\varphi$ ))  $\vee$ 
   (( $\alpha = \tau$ )  $\wedge$  ( $\exists p'. p \mapsto_\tau p' \wedge p' \models$ ?[Some X]  $\varphi$ ))  $\vee$ 
   ((idle p X)  $\wedge$  (p  $\models$ ?[None] (HMLt_poss  $\alpha \varphi$ ))))⟩
| ⟨(p  $\models$ ?[Some X] (HMLt_time Y  $\varphi$ )) = (X  $\subseteq$  visible_actions  $\wedge$ 
  ((idle p X)  $\wedge$  (p  $\models$ ?[None] (HMLt_time Y  $\varphi$ ))))⟩
⟨proof⟩

```

The well-founded relation used for the termination proof of the satisfaction function is considerably more difficult due to the last line of the definition containing the same formula on both sides of the implication (as opposed to the other lines of the definition, where the premises only contain subformulas of the formula in the conclusion). We define two relations  $R$  and  $S$ , prove their well-foundedness separately, and show that  $R \circ S \subseteq R$  (where  $\circ$  is relation composition), yielding that the union of  $R$  and  $S$  is well-founded using the theorem  $\llbracket \text{wf } R; \text{wf } S; R \circ S \subseteq R \rrbracket \implies \text{wf } (R \cup S)$ . Further details are omitted from the thesis document.

**termination** HMLt\_sat *⟨proof⟩*

We can now introduce the more readable notation (more closely corresponding to the notation in [vG20]) through abbreviations.

```

abbreviation HMLt_sat_triggered :: ⟨'s $\Rightarrow$ ('a)HMLt_formula  $\Rightarrow$  bool⟩
  (_  $\models$  _ [50, 50] 50)
  where ⟨p  $\models$   $\varphi \equiv p \models$ ?[None]  $\varphi$ ⟩
abbreviation HMLt_sat_stable :: ⟨'s $\Rightarrow$ 'a set $\Rightarrow$ ('a)HMLt_formula  $\Rightarrow$  bool⟩
  (_  $\models$  [_] _ [70, 70, 70] 80)
  where ⟨p  $\models$  [X]  $\varphi \equiv p \models$ ?[Some X]  $\varphi$ ⟩

```

Lastly, we show (by induction over  $\varphi$ ) that the function `HMLt_sat` does indeed satisfy the last clause of the original definition.

**proposition**

```

assumes
  ⟨X  $\subseteq$  visible_actions⟩
  ⟨idle p X⟩
  ⟨p  $\models$   $\varphi$ ⟩
shows
  ⟨p  $\models$  [X]  $\varphi$ ⟩
⟨proof⟩

```

As the last clause of van Glabbeek's definition is the main disparity to the function definition of `HMLt_sat`, this proposition gives confidence that the function does indeed formalise the original definition.

**end** — of context `lts_timeout`



## Chapter 3

# The Reductions

In [vG20], various characterisations of reactive bisimilarity are presented, three of which we have seen in previous sections (SRBs, GSRBs, and the modal characterisation). Another one introduces *environment operators*  $\theta_X$ , which ‘place a process in [a *stable*] environment that allows exactly the actions in  $X$  to occur’ [vG20, section 4]. The precise semantics are given by structural operational rules, e.g.:  $p \xrightarrow{\tau} p' \implies \theta_X(p) \xrightarrow{\tau} \theta_X(p')$ . For the characterisation of reactive bisimilarity, the definition of another kind of relations, namely *time-out bisimulations*, is required.

This inspired me to come up with a mapping (from  $\text{LTS}_t$ s to LTSs) that explicitly models the entire behaviour of the environment and its interaction with the reactive system. Concretely, the resulting LTS will contain a state for each state of the original  $\text{LTS}_t$  in every possible environment (including indeterminate ones). Therefore, the resulting LTS will not be a model of a reactive system, but of the closed system consisting of the original underlying system and its environment.

This approach basically treats the environment as an unknown process placed in parallel with the system; this has also been suggested by van Glabbeek in [vG20, section 2]. There, however, the action  $t$  must still be treated as a special action with special semantics. For the reduction presented in the next sections, the entire semantics of  $\text{LTS}_t$ s will be incorporated in the mapping, where all actions are then treated equally, and so that two processes of an  $\text{LTS}_t$  are strongly reactive bisimilar iff their corresponding processes in the mapped LTS are strongly bisimilar. The mapping will be presented in [section 3.1](#) and the reduction established in [section 3.2](#).

As a natural consequence, a reduction for the satisfaction of  $\text{HML}_t$  formulas can be given as well: in [section 3.3](#), I will present a mapping from  $\text{HML}_t$  formulas to HML formulas such that, as we will see in [section 3.4](#), a mapped formula holds in a process of a mapped LTS iff the original formula holds in the corresponding process of the original  $\text{LTS}_t$ .

### 3.1 A Mapping for Transition Systems

---

Let  $\mathbb{T} = (Proc, Act, \rightarrow)$  be an  $LTS_t$ . Let  $A = Act \setminus \{\tau, t\}$ .

In reference to van Glabbeek's  $\theta_X$ -operators, I introduce a family of operators  $\vartheta_X$  with similar but not identical semantics. Additionally, I introduce the operator  $\vartheta$  that places a process in an indeterminate environment.

Furthermore, I introduce a family of special actions  $\varepsilon_X$  for  $X \subseteq A$  that represent a triggered environment stabilising into an environment  $X$ , as well as a special action  $t_\varepsilon$  that represents a time-out of the environment.

We assume that  $t_\varepsilon \notin Act$  and  $\forall X \subseteq A. \varepsilon_X \notin Act$ .

Then we define  $\mathbb{T}_\vartheta = (Proc_\vartheta, Act_\vartheta, \rightarrow_\vartheta)$  with

$$\begin{aligned} Proc_\vartheta &= \{\vartheta(p) \mid p \in Proc\} \cup \{\vartheta_X(p) \mid p \in Proc \wedge X \subseteq A\}, \\ Act_\vartheta &= Act \cup \{t_\varepsilon\} \cup \{\varepsilon_X \mid X \subseteq A\}, \end{aligned}$$


and  $\rightarrow_\vartheta$  defined by the following rules:

$$\begin{aligned} (1) \quad & \frac{p \xrightarrow{\tau} p'}{\vartheta(p) \xrightarrow{\tau} \vartheta(p')} & (2) \quad \frac{}{\vartheta(p) \xrightarrow{\varepsilon_X} \vartheta_X(p)} \quad X \subseteq A \\ (3) \quad & \frac{p \xrightarrow{a} p'}{\vartheta_X(p) \xrightarrow{a} \vartheta(p')} \quad a \in X & (4) \quad \frac{p \xrightarrow{\tau} p'}{\vartheta_X(p) \xrightarrow{\tau} \vartheta_X(p')} \\ (5) \quad & \frac{p \not\xrightarrow{\alpha} \text{ for all } \alpha \in X \cup \{\tau\}}{\vartheta_X(p) \xrightarrow{t_\varepsilon} \vartheta(p)} & (6) \quad \frac{p \not\xrightarrow{\alpha} \text{ for all } \alpha \in X \cup \{\tau\} \quad p \xrightarrow{t} p'}{\vartheta_X(p) \xrightarrow{t} \vartheta_X(p')} \end{aligned}$$

These rules mirror the clauses of the definition of SRBs (cmp. [section 2.5](#)):

1.  $\tau$ -transitions can be performed regardless of the environment,
2. indeterminate environments can stabilise into arbitrary stable environments  $X$  for  $X \subseteq A$ ,
3. facilitated visible transitions can be performed and can trigger a change in the environment,
4.  $\tau$ -transitions cannot be observed by the environment and hence cannot trigger a change,
5. if the underlying system is idle, the environment may time-out and turn into an indeterminate/triggered environment,
6. if the underlying system is idle and has a  $t$ -transition, the transition may be performed and is not observable by the environment.

**Example** The  $LTS_t$  on the left (with  $Act = \{a, \tau, t\}$ ) gets mapped to the LTS on the right. States that have no incoming or outgoing transitions other than  $\varepsilon_X$  or  $t_\varepsilon$  are omitted. Note how  $\vartheta(p_4)$  is not reachable from  $\vartheta(p)$ .




---

Isabelle

---

### Formalising $Proc_\vartheta$ and $Act_\vartheta$

We specify another locale based on `lts_timeout`, where the aforementioned special actions and operators are considered; we call it `lts_timeout_mappable`. Since  $Proc \cap Proc_\vartheta = \emptyset$ , we introduce a new type variable 'ss for  $Proc_\vartheta$ ; we use 'a for both  $Act$  and  $Act_\vartheta$ . We formalise the family of special actions  $\varepsilon_X$  as a mapping  $\varepsilon[_] :: 'a \text{ set} \Rightarrow 'a$ , and the environment operators  $\vartheta/\vartheta_X$  as a single mapping  $\vartheta?[_](_) :: 'a \text{ set option} \Rightarrow 's \Rightarrow 'ss$ .

As for `lts_timeout` in [section 2.4](#), we require that all special actions are distinct, formalised by the first set of assumptions `distinctness_special_actions`.

As an operator, the term  $\vartheta_X(p)$  simply refers to the state  $p$  in an environment  $X$ ; when understood as a mapping, we have to be more careful, since  $\vartheta?[Some\ X](p)$  is now itself a state. Specifically, we have to assume that  $\vartheta?[_](_)$  is injective (when restricted to domains where  $X \subseteq \text{visible\_actions}$ , because  $\vartheta_X$  is only defined for those  $X \subseteq A$ ). Otherwise, we might have  $\vartheta?[None](p) = \vartheta?[None](q)$  for  $p \neq q$ , which is problematic if e.g.  $p$  has a  $\tau$ -transition, but  $q$  does not. The restricted injectivity of  $\vartheta?[_](_)$  is formalised as the set of assumptions `injectivity_theta`.

The same is required for the mapping  $\varepsilon[_]$ , as formalised in the last clause of the set of assumptions `distinctness_special_actions` (the restricted injectivity of  $\varepsilon[_]$  is part of the requirement that all special actions must be distinct). Again, we only require injectivity for the mapping restricted to the domain `visible_actions`. If we required that  $\varepsilon[_] :: 'a \text{ set} \Rightarrow 'a$  were

injective over its entire domain 'a set, we would run into problems, since such a function cannot exist by Cantor's theorem.

That such mappings exist is intuitively clear, whence there were no ambiguities when defining them as operators in the prosaic/mathematical section above. Formalising these mappings in HOL, however, is not so straightforward: as operators, we assume that they are only defined for certain parameters; in HOL, every mapping must be total. For now, we simply assume that such total functions that formalise these operators exist. Doing this significantly improves the readability of following sections, since we must only consider the relevant properties of the mappings given by the assumptions. In [appendix C](#), I give examples for these mappings and show that, together with these, any `lts_timeout` can be interpreted as an `lts_timeout_mappable`, i.e. every  $\text{LTS}_t \text{ T}$  can be mapped to an  $\text{LTS } T_\theta$ .

Lastly, we formalise our requirements  $t_\varepsilon \notin \text{Act}$  and  $\forall X \subseteq A. \varepsilon_X \notin \text{Act}$  as the last set of assumptions `no_epsilon_in_tran`. Technically, these assumptions only state that the  $\varepsilon$ -actions do not label any transition of  $T$ . However, we can assume that  $\text{Act} = \{\alpha \mid \exists p, p'. p \xrightarrow{\alpha} p'\}$ , since actions that do not label any transitions are not relevant to the behaviour of an LTS.

```

locale lts_timeout_mappable = lts_timeout tran  $\tau$  t
  for tran :: 's  $\Rightarrow$  'a  $\Rightarrow$  's  $\Rightarrow$  bool
    ( $\_ \mapsto \_$  [70, 70, 70] 80)
  and  $\tau$  :: 'a
  and t :: 'a +
  fixes t_ε :: 'a
  and stabilise :: ('a set  $\Rightarrow$  'a)
    ( $\langle \varepsilon[\_] \rangle$ )
  and in_env :: ('a set option  $\Rightarrow$  's  $\Rightarrow$  'ss)
    ( $\langle \vartheta?[\_] '(\_) \rangle$ )
  assumes
    distinctness_special_actions:
      ( $\tau \neq t$ ) ( $\tau \neq t_\varepsilon$ ) ( $t \neq t_\varepsilon$ )
      ( $\langle \varepsilon[X] \neq \tau \rangle$ ) ( $\langle \varepsilon[X] \neq t \rangle$ ) ( $\langle \varepsilon[X] \neq t_\varepsilon \rangle$ )
      ( $X \subseteq \text{visible\_actions} \implies \varepsilon[X] = \varepsilon[Y] \implies X = Y$ )
  and

    injectivity_theta:
      ( $\langle \vartheta?[\text{None}] (p) \neq \vartheta?[\text{Some } X] (q) \rangle$ )
      ( $\langle (\vartheta?[\text{None}] (p) = \vartheta?[\text{None}] (q)) \longrightarrow p = q \rangle$ )
      ( $X \subseteq \text{visible\_actions} \implies$ 
        ( $\langle \vartheta?[\text{Some } X] (p) = \vartheta?[\text{Some } Y] (q) \rangle \longrightarrow X = Y \wedge p = q$ )
      )
  and

    no_epsilon_in_tran:
      ( $\neg p \mapsto \varepsilon[X] q$ )
      ( $\neg p \mapsto t_\varepsilon q$ )
begin

```

We can now define abbreviations with notations that correspond more closely to our operators defined above.

```

abbreviation triggered_env :: ('s  $\Rightarrow$  'ss)
  ( $\vartheta$ '(' _'))
  where  $\vartheta$ (p)  $\equiv$   $\vartheta$ ?[None](p)
abbreviation stable_env :: ('a set  $\Rightarrow$  's  $\Rightarrow$  'ss)
  ( $\vartheta$ [_]'(' _'))
  where  $\vartheta$ [X](p)  $\equiv$   $\vartheta$ ?[Some X](p)

```

### Formalising $\rightarrow_{\vartheta}$

We formalise the transition relation of our mapping, given above by the structural operational rules, as a function `tran_theta`.<sup>1</sup>

We use the `inductive` command, because it allows us to define separate clauses (as opposed to the `definition` command). Technically speaking, however, this inductive definition only has base cases, since none of the premises involves  $\mapsto_{\vartheta}$ .

It should be easy to see that the clauses below correspond directly to the rules above. Like in previous sections, we have to take extra care to handle the requirement  $X \subseteq \text{visible\_actions}$ .

```

inductive tran_theta :: ('ss  $\Rightarrow$  'a  $\Rightarrow$  'ss  $\Rightarrow$  bool)
  ( $\_ \mapsto_{\vartheta} \_$ ) [70, 70, 70] 70)
  where
    triggered_tau:
       $\langle p \mapsto_{\tau} q \implies \vartheta(p) \mapsto_{\tau} \vartheta(q) \rangle$ 
    | env_stabilise:  $\langle X \subseteq \text{visible\_actions} \implies$ 
       $\vartheta(p) \mapsto_{\varepsilon[X]} \vartheta[X](p) \rangle$ 
    | tran_visible:  $\langle X \subseteq \text{visible\_actions} \implies$ 
       $a \in X \implies p \mapsto_a q \implies \vartheta[X](p) \mapsto_a \vartheta(q) \rangle$ 
    | stable_tau:  $\langle X \subseteq \text{visible\_actions} \implies$ 
       $p \mapsto_{\tau} q \implies \vartheta[X](p) \mapsto_{\tau} \vartheta[X](q) \rangle$ 
    | env_timeout:  $\langle X \subseteq \text{visible\_actions} \implies$ 
       $\text{idle } p \ X \implies \vartheta[X](p) \mapsto_{t_{\varepsilon}} \vartheta(p) \rangle$ 
    | sys_timeout:  $\langle X \subseteq \text{visible\_actions} \implies$ 
       $\text{idle } p \ X \implies p \mapsto_t q \implies \vartheta[X](p) \mapsto_t \vartheta[X](q) \rangle$ 

```

### Note on Metavariable usage

If not referenced directly by  $\vartheta(p)$  or  $\vartheta_X(p)$ , arbitrary states of a mapped LTS range over  $P, Q, P', Q', \dots$ , where  $P$  and  $P'$  are used for states connected by some transition (i.e.  $P \xrightarrow{\alpha}_{\vartheta} P'$ ), whereas  $P$  and  $Q$  are used for states possibly related by some equivalence (e.g.  $P \leftrightarrow_r Q$ ).

<sup>1</sup>We use the notation  $\_ \mapsto_{\vartheta} \_$  instead of the more obvious  $\_ \mapsto_{\vartheta} \_$  simply because of better readability.

### Generation Lemmas

Lastly, we derive a set of generation lemmas, i.e. lemmas that allow us to reason backwards: if we know  $P \mapsto^\vartheta \alpha P'$  and some other information about  $P$  and/or  $\alpha$ , we can deduce some information about the other variables as well as the transitions of the original  $\text{LTS}_t$ .

**lemma generation\_triggered\_transitions:**

**assumes**  $\langle \vartheta(p) \mapsto^\vartheta \alpha P' \rangle$   
**shows**  $\langle (\exists X. \alpha = \varepsilon[X] \wedge P' = \vartheta[X](p) \wedge X \subseteq \text{visible\_actions})$   
 $\vee (\alpha = \tau \wedge (\exists p'. p \mapsto_\tau p')) \rangle$   
*⟨proof⟩*

**lemma generation\_stable\_transitions:**

**assumes**  $\langle \vartheta[X](p) \mapsto^\vartheta \alpha P' \rangle$   
**shows**  $\langle \alpha = t_\varepsilon \vee (\exists p'. p \mapsto_\alpha p' \wedge (\alpha \in X \vee \alpha = \tau \vee \alpha = t)) \rangle$   
*⟨proof⟩*

**lemma generation\_triggered\_tau:**

**assumes**  $\langle \vartheta(p) \mapsto^\vartheta \tau P' \rangle$   
**shows**  $\langle \exists p'. P' = \vartheta(p') \wedge p \mapsto_\tau p' \rangle$   
*⟨proof⟩*

**lemma generation\_env\_stabilise:**

**assumes**  $\langle P \mapsto^\vartheta \varepsilon[X] P' \rangle$   
**shows**  $\langle \exists p. P = \vartheta(p) \wedge P' = \vartheta[X](p) \rangle$   
*⟨proof⟩*

**lemma generation\_tran\_visible:**

**assumes**  $\langle \vartheta[X](p) \mapsto^\vartheta a P' \rangle$   $\langle a \in \text{visible\_actions} \rangle$   
**shows**  $\langle a \in X \wedge (\exists p'. P' = \vartheta(p') \wedge p \mapsto_a p') \rangle$   
*⟨proof⟩*

**lemma generation\_stable\_tau:**

**assumes**  $\langle \vartheta[X](p) \mapsto^\vartheta \tau P' \rangle$   
**shows**  $\langle \exists p'. P' = \vartheta[X](p') \wedge p \mapsto_\tau p' \rangle$   
*⟨proof⟩*

**lemma generation\_env\_timeout:**

**assumes**  $\langle \vartheta[X](p) \mapsto^\vartheta t_\varepsilon P' \rangle$   
**shows**  $\langle P' = \vartheta(p) \wedge \text{idle } p \ X \rangle$   
*⟨proof⟩*

**lemma generation\_sys\_timeout:**

**assumes**  $\langle \vartheta[X](p) \mapsto^\vartheta t P' \rangle$   
**shows**  $\langle \exists p'. P' = \vartheta[X](p') \wedge \text{idle } p \ X \wedge p \mapsto_t p' \rangle$   
*⟨proof⟩*

**end** — of locale `lts_timeout_mappable`

### 3.2 Reduction of Bisimilarity

---

The result of this section will be that two processes  $p$  and  $q$  of an  $\text{LTS}_t \mathbb{T}$  are strongly reactive bisimilar (strongly  $X$ -bisimilar) iff the corresponding processes  $\vartheta(p)$  and  $\vartheta(q)$  ( $\vartheta_X(p)$  and  $\vartheta_X(q)$ ) of  $\mathbb{T}_\vartheta$  are strongly bisimilar.

We show the  $\implies$ -direction first. For an SRB  $\mathcal{R}$ , let

$$\mathcal{S} = \{(\vartheta(p), \vartheta(q)) \mid (p, q) \in \mathcal{R}\} \cup \{(\vartheta_X(p), \vartheta_X(q)) \mid (p, X, q) \in \mathcal{R}\}.$$

We can prove that  $\mathcal{S}$  is an SB, by showing that the mapping satisfies the clauses of the definition of SBs, using the fact that  $\mathcal{R}$  is an SRB as well as the rules and generation lemmas for  $\rightarrow_\vartheta$ . Hence, the existence of an SRB  $\mathcal{R}$  with  $(p, q) \in \mathcal{R}$  implies the existence of an SB  $\mathcal{S}$  with  $(\vartheta(p), \vartheta(q)) \in \mathcal{S}$  (and similarly for  $\vartheta_X$ ), so strong reactive/ $X$ -bisimilarity in  $\mathbb{T}$  implies strong bisimilarity in  $\mathbb{T}_\vartheta$ .

Next, we show the  $\impliedby$ -direction. Let

$$\mathcal{R} = \{(p, q) \mid \vartheta(p) \leftrightarrow \vartheta(q)\} \cup \{(p, X, q) \mid \vartheta_X(p) \leftrightarrow \vartheta_X(q)\}.$$

We can prove that  $\mathcal{R}$  is an SRB, again, by showing that all clauses of the definition are satisfied. Hence, strong bisimilarity of  $\vartheta(p)$  and  $\vartheta(q)$  implies the existence of an SRB  $\mathcal{R}$  with  $(p, q) \in \mathcal{R}$  (and similarly for  $\vartheta_X$ ), so strong bisimilarity in  $\mathbb{T}_\vartheta$  implies strong reactive/ $X$ -bisimilarity in  $\mathbb{T}$ .

Thus, we have that strong reactive/ $X$ -bisimilarity in  $\mathbb{T}$  corresponds to strong bisimilarity in  $\mathbb{T}_\vartheta$ .

---

#### Isabelle

---

We begin by *interpreting* our transition mapping `tran_theta` as an `lts` and call it `lts_theta`. Therefore, we are handling two separate LTSs: the  $\text{LTS}_t \mathbb{T}$  given by the local context `lts_timeout_mappable`, and the  $\text{LTS} \mathbb{T}_\vartheta$  given by the interpretation `lts_theta`. When referring to definitions involving the transition relation of `lts_theta`, we have to prefix them, e.g. `lts_theta.SB` for the definition of strong bisimulations using  $\mapsto^\vartheta$  instead of  $\mapsto$ .

By default, `interpretations` do not import special notation, so we reassign strong bisimilarity notation  $\leftrightarrow$  to `lts_theta`, since we do not care about strong bisimilarity in  $\mathbb{T}$ .

```
context lts_timeout_mappable begin
```

```
interpretation lts_theta: lts tran_theta <proof>
no_notation local.strongly_bisimilar (<_ <-> _> [70, 70] 70)
notation lts_theta.strongly_bisimilar (<_ <-> _> [70, 70] 70)
```

We can now formalise the proof as described above.

If ... ( $\implies$ )

**definition** `SRB_mapping` —  $\mathcal{S}$

```

:: <('s=>'a set option=>'s => bool) => ('ss=>'ss => bool)>
where <SRB_mapping R P Q ≡
  (∃ p q. P = ∅(p) ∧ Q = ∅(q) ∧ R p None q) ∨
  (∃ p q X. P = ∅[X](p) ∧ Q = ∅[X](q) ∧ R p (Some X) q)>

```

**lemma** `SRB_mapping_is_SB`:

```

assumes <SRB R>
shows <lts_theta.SB (SRB_mapping R)> (is <lts_theta.SB ?S>)
<proof>

```

**lemma** `srby_implies_sby`:

```

assumes <p ↔r q>
shows <∅(p) ↔ ∅(q)>
<proof>

```

**lemma** `sxby_implies_sby`:

```

assumes <p ↔rX q>
shows <∅[X](p) ↔ ∅[X](q)>
<proof>

```

... and only if ( $\iff$ )

**definition** `strong_bisimilarity_mapping` —  $\mathcal{R}$

```

:: <'s=>'a set option=>'s => bool>
where <(strong_bisimilarity_mapping) p XoN q
  ≡ (XoN = None ∧ (∅(p) ↔ ∅(q))) ∨
  (∃ X. XoN = Some X ∧ X ⊆ visible_actions ∧
   ∅[X](p) ↔ ∅[X](q))>

```

**lemma** `strong_bisimilarity_mapping_is_SRB`:

```

shows <SRB strong_bisimilarity_mapping> (is <SRB ?R>)
<proof>

```

**lemma** `sby_implies_srby`:

```

assumes <∅(p) ↔ ∅(q)>
shows <p ↔r q>
<proof>

```

**lemma** `sby_implies_sxby`:

```

assumes <∅[X](p) ↔ ∅[X](q)> <X ⊆ visible_actions>
shows <p ↔rX q>
<proof>

```

We need to include the assumption  $X \subseteq \text{visible\_actions}$ , since for  $\neg X \subseteq \text{visible\_actions}$ ,  $\emptyset[X](p)$  and  $\emptyset[X](q)$  might be identical (since we do not require injectivity for that subset of the domain), so  $\emptyset[X](p) \leftrightarrow \emptyset[X](q)$  would be true, whereas  $p \leftrightarrow_r^X q$  would be false (since  $X \subseteq \text{visible\_actions}$  is part of the definition of SRBs).



**Iff** ( $\iff$ )

**theorem** `strongly_reactive_bisim_iff_triggered_strongly_bisim`:  
 shows  $\langle p \leftrightarrow_r q \iff \vartheta(p) \leftrightarrow \vartheta(q) \rangle$   
 using `sby_implies_srby srby_implies_sby` by fast

**theorem** `strongly_X_bisim_iff_stable_strongly_bisim`:  
 assumes  $\langle X \subseteq \text{visible\_actions} \rangle$   
 shows  $\langle p \leftrightarrow_r^X q \iff \vartheta[X](p) \leftrightarrow \vartheta[X](q) \rangle$   
 using `sxby_implies_sby sby_implies_sxby` `assms` by fast

**end** — of context `lts_timeout_mappable`

### 3.3 A Mapping for Formulas

---

We will now introduce a mapping  $\sigma(\cdot)$  that maps formulas of  $\text{HML}_t$  to formulas of  $\text{HML}$ , in the context of the process mapping from [section 3.1](#), such that  $\vartheta(p)$  satisfies  $\sigma(\varphi)$  iff  $p$  satisfies  $\varphi$ .

Again, we have  $\mathbb{T} = (\text{Proc}, \text{Act}, \rightarrow)$  and  $\mathbb{T}_\vartheta = (\text{Proc}_\vartheta, \text{Act}_\vartheta, \rightarrow_\vartheta)$  as defined in [section 3.1](#), with  $A = \text{Act} \setminus \{\tau, t\}$ , and we assume that  $t_\varepsilon \notin \text{Act}$  and  $\forall X \subseteq A. \varepsilon_X \notin \text{Act}$ .

Let  $\sigma : (\text{HML}_t \text{ formulas}) \longrightarrow (\text{HML formulas})$  be recursively defined by

$$\begin{aligned}
 \sigma(\bigwedge_{i \in I} \varphi_i) &= \bigwedge_{i \in I} \sigma(\varphi_i) \\
 \sigma(\neg \varphi) &= \neg \sigma(\varphi) \\
 \sigma(\langle \tau \rangle \varphi) &= \langle \tau \rangle \sigma(\varphi) \\
 \sigma(\langle \alpha \rangle \varphi) &= \langle \alpha \rangle \sigma(\varphi) \vee \\
 &\quad \langle \varepsilon_A \rangle \langle \alpha \rangle \sigma(\varphi) \vee \\
 &\quad \langle t_\varepsilon \rangle \langle \varepsilon_A \rangle \langle \alpha \rangle \sigma(\varphi) && \text{if } \alpha \in A \\
 \sigma(\langle \alpha \rangle \varphi) &= \text{ff} && \text{if } \alpha \notin A \cup \{\tau\} \\
 \sigma(\langle X \rangle \varphi) &= \langle \varepsilon_X \rangle \langle t \rangle \sigma(\varphi) \vee \\
 &\quad \langle t_\varepsilon \rangle \langle \varepsilon_X \rangle \langle t \rangle \sigma(\varphi) && \text{if } X \subseteq A \\
 \sigma(\langle X \rangle \varphi) &= \text{ff} && \text{if } X \not\subseteq A
 \end{aligned}$$

This mapping simply expresses the time-out semantics given by the satisfaction relations of  $\text{HML}_t$  ([section 2.6](#)) in terms of ordinary  $\text{HML}$  evaluated on our mapped LTS  $\mathbb{T}_\vartheta$ . The disjunctive clauses compensate for the additional environment transitions ( $\varepsilon$ -actions) that are not present in  $\mathbb{T}$ .

---

Isabelle

---

The implementation of the mapping in Isabelle is rather straightforward, although some details might not be obvious:

$\text{cimage } (\lambda \varphi. \sigma(\varphi)) \Phi$  is the image of the countable set  $\Phi$  under the function  $\lambda \varphi. \sigma(\varphi)$ , so it corresponds to  $\{\sigma(\varphi) \mid \varphi \in \Phi\}$  for countable  $\Phi$ .

$\alpha \neq \tau \wedge \alpha \neq \mathbf{t} \wedge \alpha \neq \mathbf{t}_\varepsilon \wedge (\forall X. \alpha \neq \varepsilon[X])$  corresponds to  $\alpha \in A$  with our assumption about there being no  $\varepsilon$ -actions in  $Act$ . Similarly,  $\alpha = \mathbf{t} \vee \alpha = \mathbf{t}_\varepsilon \vee \alpha = \varepsilon[X]$  corresponds to  $\alpha \notin A \cup \{\tau\}$ .

**context** `lts_timeout_mappable` **begin**

```
function HMT_mapping :: (('a)HMLt_formula  $\Rightarrow$  ('a)HML_formula)
  ( $\langle \sigma'(\_) \rangle$ )
where
   $\langle \sigma(\text{HMLt\_conj } \Phi) = \text{HML\_conj } (\text{cimage } (\lambda \varphi. \sigma(\varphi)) \Phi) \rangle$ 
  |  $\langle \sigma(\text{HMLt\_neg } \varphi) = \text{HML\_neg } \sigma(\varphi) \rangle$ 
  |  $\langle \alpha = \tau \implies$ 
     $\sigma(\text{HMLt\_poss } \alpha \varphi) = \text{HML\_poss } \alpha \sigma(\varphi) \rangle$ 
  |  $\langle \alpha \neq \tau \wedge \alpha \neq \mathbf{t} \wedge \alpha \neq \mathbf{t}_\varepsilon \wedge (\forall X. \alpha \neq \varepsilon[X]) \implies$ 
     $\sigma(\text{HMLt\_poss } \alpha \varphi) = \text{HML\_disj } (\text{acset } \{$ 
       $\text{HML\_poss } \alpha \sigma(\varphi),$ 
       $\text{HML\_poss } \varepsilon[\text{visible\_actions}] (\text{HML\_poss } \alpha \sigma(\varphi)),$ 
       $\text{HML\_poss } \mathbf{t}_\varepsilon (\text{HML\_poss } \varepsilon[\text{visible\_actions}] (\text{HML\_poss } \alpha \sigma(\varphi)))$ 
     $\}) \rangle$ 
  |  $\langle \alpha = \mathbf{t} \vee \alpha = \mathbf{t}_\varepsilon \vee \alpha = \varepsilon[X] \implies$ 
     $\sigma(\text{HMLt\_poss } \alpha \varphi) = \text{HML\_false} \rangle$ 
  |  $\langle X \subseteq \text{visible\_actions} \implies$ 
     $\sigma(\text{HMLt\_time } X \varphi) = \text{HML\_disj } (\text{acset } \{$ 
       $\text{HML\_poss } \varepsilon[X] (\text{HML\_poss } \mathbf{t} \sigma(\varphi)),$ 
       $\text{HML\_poss } \mathbf{t}_\varepsilon (\text{HML\_poss } \varepsilon[X] (\text{HML\_poss } \mathbf{t} \sigma(\varphi)))$ 
     $\}) \rangle$ 
  |  $\langle \neg X \subseteq \text{visible\_actions} \implies$ 
     $\sigma(\text{HMLt\_time } X \varphi) = \text{HML\_false} \rangle$ 
   $\langle \text{proof} \rangle$ 
```

Again, we show that the function terminates using a well-founded relation.

```
inductive_set sigma_wf_rel :: (('a)HMLt_formula) rel)
where
   $\langle \varphi \in_c \Phi \implies (\varphi, \text{HMLt\_conj } \Phi) \in \text{sigma\_wf\_rel} \rangle$ 
  |  $\langle (\varphi, \text{HMLt\_neg } \varphi) \in \text{sigma\_wf\_rel} \rangle$ 
  |  $\langle (\varphi, \text{HMLt\_poss } \alpha \varphi) \in \text{sigma\_wf\_rel} \rangle$ 
  |  $\langle (\varphi, \text{HMLt\_time } X \varphi) \in \text{sigma\_wf\_rel} \rangle$ 
```

**termination** HMT\_mapping  $\langle \text{proof} \rangle$

**end** — of context `lts_timeout_mappable`

### 3.4 Reduction of Formula Satisfaction

---

We will show that, for a process  $p$  of an  $\text{LTS}_t$  and a formula  $\varphi$  of  $\text{HML}_t$ , we have  $p \models \varphi \iff \vartheta(p) \models \sigma(\varphi)$  and  $p \models_X \varphi \iff \vartheta_X(p) \models \sigma(\varphi)$ . The proof is rather straightforward: we use induction over  $\text{HML}_t$  formulas and show that, for each case, the semantics given by van Glabbeek's satisfaction relations and those given by the mappings  $\sigma$  and  $\vartheta/\vartheta_X$  coincide. Due to the relative complexity of the mapping and the satisfaction relations, the proof is quite tedious, however.

---

Isabelle

---

Similarly to the formalisations in [section 3.2](#), we begin by interpreting our transition mapping `tran_theta` as an `lts` and reassigning notation appropriately (we only care about HML formula satisfaction for  $\mathbb{T}_\vartheta$ , not  $\mathbb{T}$ ).

```
context lts_timeout_mappable begin
```

```
interpretation lts_theta: lts tran_theta <proof>
no_notation local.HML_sat (_ ⊨ _ [50, 50] 50)
notation lts_theta.HML_sat (_ ⊨ _ [50, 50] 50)
```

We show  $p \models_{\text{?}[XoN]} \varphi \iff \vartheta_{\text{?}[XoN]}(p) \models \sigma(\varphi)$  by induction over  $\varphi$ . By using those terms for formula satisfaction and process mappings that handle both triggered and stable environments, we can handle both situations simultaneously, which is required due to the interdependence of  $\models$  and  $\models_X$ . However, this requires us to consider four cases (each combination of  $\models_{\text{?}[XoN_1]}$  and  $\vartheta_{\text{?}[XoN_2]}$  for  $XoN_1, XoN_2 \in \{\text{None}, \text{Some } X\}^2$ ) per inductive case for  $\varphi$ . Together with the many disjunctive clauses in the mapping, a large number of cases needs to be considered, leading to a proof spanning roughly 350 lines of Isabelle code.

```
lemma HMLt_sat_iff_HML_sat:
  assumes <XoN = None ∨ (XoN = (Some X) ∧ X ⊆ visible_actions)>
  shows <p ⊨_{?}[XoN] ϕ ⟷ ϑ_{?}[XoN](p) ⊨ σ(ϕ)>
  <proof>
```

Theorems using nicer notation are immediate consequences of this lemma.

```
theorem HMLt_sat_triggered_iff_triggered_env_HML_sat:
  shows <p ⊨ ϕ ⟷ ϑ(p) ⊨ σ(ϕ)>
  using HMLt_sat_iff_HML_sat by blast
theorem HMLt_sat_stable_iff_stable_env_HML_sat:
  assumes <X ⊆ visible_actions>
  shows <p ⊨[X] ϕ ⟷ ϑ[X](p) ⊨ σ(ϕ)>
  using HMLt_sat_iff_HML_sat assms by blast
```

```
end — of context lts_timeout_mappable
```

---

<sup>2</sup>Once again, we only consider cases where  $X \subseteq \text{visible\_actions}$ .

## Chapter 4

# Discussion

We have shown that checking strong reactive/ $X$ -bisimilarity (in an  $LTS_t$ ) is reducible to checking strong bisimilarity. This result may be useful in the context of automated tools for checking equivalences on LTSs. Since, the mapping creates a state for every subset of the visible actions  $A$ , for each original state, plus another triggered state (i.e.  $|Proc_\theta| = |Proc| \cdot (1 + 2^{|A|})$ ), checking reactive bisimilarity by using the mapping would be exponentially harder (in the worst case) than simply checking ordinary bisimilarity. However, at least for SRBs, the size of the relations also grows exponentially with the number of visible actions (due to the clause  $(p, q) \in \mathcal{R} \implies (p, X, q) \in \mathcal{R}$  for  $X \subseteq A$ ), so a naïve implementation using SRBs would not necessarily be more efficient. Below, I propose an optimisation that significantly reduces the number of states for a large number of LTSs.

If two processes of a mapped LTS are not strongly bisimilar, an automated tool might produce a distinguishing HML formula. Due to the clear semantics of the actions introduced as part of my mapping, such formulas would be easy to interpret (to understand which sequence of actions distinguishes the two processes).

As mentioned previously, the mapped LTS represents the closed system consisting of the original reactive system and its environment. Hence, the reduction does in no way challenge the semantic value offered by  $LTS_t$ s, e.g. for protocol specifications. Rather, when shown as a graph, the mapping might complement such specifications by offering a useful view that explicitly shows the specified system in all possible environments. In a mapped LTS, for example, it is easy to find states that are unreachable, or reachable only in certain environments, whereas the reachability of states in an  $LTS_t$  may not be directly obvious, as we saw in the example on [page 35](#). Admittedly, the mapping gets very crowded even for small  $LTS_t$ s; on a local level, however, the explicitness of the mapping may be useful. Lastly, it might be helpful simply for understanding  $LTS_t$  semantics.

The formula mapping is probably less useful in that regard, due to the large number of disjunctions. It might, however, also be useful in the context of automated tools.

### Possible Optimisations

Let  $\mathcal{I}^*(p)$  be the set of visible actions that can be encountered as initial actions after arbitrary sequences of  $\tau$ - and  $t$ -transitions starting at  $p$ .

More concretely, for  $n \in \mathbb{N}$ , let

$$\begin{aligned} p_0 \xrightarrow{X}^* p_n &: \Leftrightarrow \exists p_1, \dots, p_{n-1}. \forall i \in [0, n-1]. \exists \alpha \in X. p_i \xrightarrow{\alpha} p_{i+1}, \\ \text{reach}(p, X) &:= \{p' \mid p \xrightarrow{X}^* p'\}, \\ \mathcal{I}^*(p) &:= \bigcup_{p' \in \text{reach}(p, \{\tau, t\})} \mathcal{I}(p'). \end{aligned}$$

Then<sup>1</sup> we can modify the second rule of the process mapping (from [section 3.1](#)) by changing the side condition from  $X \subseteq A$  to  $X \subseteq \mathcal{I}^*(p)$ , yielding:

$$(2) \frac{}{\vartheta(p) \xrightarrow{\varepsilon_X}_{\vartheta} \vartheta_X(p)} X \subseteq \mathcal{I}^*(p).$$

This way, we only include environment stabilisations that are relevant for the current process: all transitions other than  $\tau$  and  $t$  will always trigger a change in the environment; hence, after having stabilised, the actions in  $\mathcal{I}^*(p)$  are the only ones the process  $p$  could ever perform before triggering the environment.

In the worst case, the number of mapping states is still exponential in the size of the alphabet, i.e.  $|\text{Proc}_{\vartheta}| = \mathcal{O}(|\text{Proc}| \cdot 2^{|Act|})$ . For a large number of LTS<sub>t</sub>s, however, this alteration would reduce the number of mapping states significantly.

For reasons of time, I did not attempt to prove the reduction with this altered mapping, but firmly believe that it is possible.

### Necessity of Special Actions

**Environment Time-Outs  $t_\varepsilon$**  It should be possible to replace the action  $t_\varepsilon$  by the normal time-out action  $t$  in the mapping. Since, in the present version, all  $t_\varepsilon$ -transitions end in a  $\vartheta(p)$ -state, where always at least an  $\varepsilon_\emptyset$ -transition can be performed, whereas all  $t$ -transitions end in a  $\vartheta_X(p)$ -state, where no  $\varepsilon_\emptyset$ -transition can ever be performed, the distinction between environment time-outs and system time-outs should be possible without distinguishing the actions  $t$  and  $t_\varepsilon$ .

---

<sup>1</sup>Note that  $p \in \text{reach}(p, X)$  for all  $p$  and  $X$ .

**Environment Stabilisations  $\varepsilon_X$**  In the first version of the mapping, I used a single stabilisation action  $s_\varepsilon$ , but got stuck trying to prove that  $\vartheta(p) \leftrightarrow \vartheta(q)$  implies  $\forall X \subseteq A. \vartheta_X(p) \leftrightarrow \vartheta_X(q)$ .

Concretely, in such a version of the mapping with only one stabilisation action, the bisimilarity property would allow us to conclude from  $\vartheta(p) \leftrightarrow \vartheta(q)$  only that  $\forall X \subseteq A. \exists Y \subseteq A. \vartheta_X(p) \leftrightarrow \vartheta_Y(q)$ . Since  $\vartheta_X(p)$  then cannot have transitions with labels in  $X \setminus Y$ , because  $\vartheta_Y(q)$  could not mirror these transitions (and also the other way around for transitions of  $\vartheta_Y(q)$  with labels in  $Y \setminus X$ ), I attempted to prove that this implies  $\forall X \subseteq A. \vartheta_X(p) \leftrightarrow \vartheta_X(q)$ . However, the fact that  $p$  and  $q$  might have  $\tau$ - or  $t$ -transitions necessitates that one also has to consider the transitions of those derivative states. Arbitrarily long sequences of  $\tau$ -/ $t$ -transitions are then the crux of this lemma. After many days of unsuccessful proof attempts, I admitted defeat and defined the family of stabilisation actions instead. Sadly, I did not manage to find a counterexample where the reduction using this simpler mapping does not work, either.

However, in the context of  $\text{HML}_t$ , there would be no obvious way to define the formula mapping for  $\sigma(\langle X \rangle \varphi)$ ; in the present version, the mapping relies on being able to use  $\varepsilon_X$  in this case (see [section 3.3](#)). Hence, I have to come believe that the  $\varepsilon_X$ -actions might indeed be required in their present form.

Furthermore, although including the environment information both in the states  $\vartheta_X(p)$  as well as the stabilisation actions  $\varepsilon_X$  may seem redundant, it might be necessary. As we discussed in [section 2.2](#), the intensional identity of the state is not ‘knowable for bisimilarity’; rather, only the observable transitions are relevant. Hence, it is plausible that the information about allowed actions is actually required to be included in the transition labels themselves, in order for the reduction to work.

## Isabelle Formalisations

The Isabelle formalisations that were done as part of this thesis have been the first formalisations of  $\text{LTS}_t$ s and related concepts.

The only Isabelle formalisation of a Hennessy-Milner logic published on the *Isabelle Archive of Formal Proofs* was presented in [\[WEP<sup>+</sup>16\]](#). The variant of HML formalised there includes *state predicates* evaluated on *nominal transition systems* (i.e. each state of the transition system is associated with a set of satisfied state predicates); the formalisations are considerably more complex than those done in this thesis.

Potential future projects requiring only ‘purely modal’ Hennessy-Milner logic might benefit from the simplicity of these formalisations. Thus, the formalisation of (simple) Hennessy-Milner logic in [section 2.3](#) or of infinitary Hennessy-Milner logic in [appendix B](#) might be useful in future research.

# Bibliography

- [AILS07] Luca Aceto, Anna Ingolfsdottir, Kim Guldstrand Larsen, and Jirí Srba. *Reactive systems: modelling, specification and verification*. Cambridge University Press, 2007. doi:[10.1017/CB09780511814105](https://doi.org/10.1017/CB09780511814105).
- [AIS11] Luca Aceto, Anna Ingolfsdottir, and Jirí Srba. The algorithmics of bisimilarity. In Davide Sangiorgi and Jan Rutten, editors, *Advanced Topics in Bisimulation and Coinduction*, Cambridge Tracts in Theoretical Computer Science, pages 100—172. Cambridge University Press, 2011. doi:[10.1017/CB09780511792588.004](https://doi.org/10.1017/CB09780511792588.004).
- [BN19] Benjamin Bisping and Uwe Nestmann. Computing coupled similarity. In Tomáš Vojnar and Lijun Zhang, editors, *Tools and Algorithms for the Construction and Analysis of Systems*, pages 244–261. Springer International Publishing, 2019. doi:[10.1007/978-3-030-17462-0\\_14](https://doi.org/10.1007/978-3-030-17462-0_14).
- [HM85] Matthew Hennessy and Robin Milner. Algebraic laws for nondeterminism and concurrency. *Journal of the ACM*, 32(1):137—161, 1985. doi:[10.1145/2455.2460](https://doi.org/10.1145/2455.2460).
- [HP85] David Harel and Amir Pnueli. On the development of reactive systems. In Krzysztof R. Apt, editor, *Logics and models of concurrent systems*, volume 13, pages 477–498. Springer Berlin Heidelberg, 1985. doi:[10.1007/978-3-642-82453-1\\_17](https://doi.org/10.1007/978-3-642-82453-1_17).
- [Kel76] Robert M. Keller. Formal verification of parallel programs. *Communications of the ACM*, 19(7):371–384, 1976. doi:[10.1145/360248.360251](https://doi.org/10.1145/360248.360251).
- [Nip21] Tobias Nipkow. Programming and proving in Isabelle/HOL, February 2021. Accessed on 07.06.2021. URL: <https://isabelle.in.tum.de/doc/prog-prove.pdf>.

- [Par81] David Park. Concurrency and automata on infinite sequences. In Peter Deussen, editor, *Theoretical Computer Science*, pages 167–183. Springer Berlin Heidelberg, 1981. doi:[10.1007/BFb0017309](https://doi.org/10.1007/BFb0017309).
- [San11] Davide Sangiorgi. Origins of bisimulation and coinduction. In Davide Sangiorgi and Jan Rutten, editors, *Advanced Topics in Bisimulation and Coinduction*, Cambridge Tracts in Theoretical Computer Science, pages 1—37. Cambridge University Press, 2011. doi:[10.1017/CB09780511792588.002](https://doi.org/10.1017/CB09780511792588.002).
- [San12] Davide Sangiorgi. *Introduction to bisimulation and coinduction*. Cambridge University Press, 2012. doi:[10.1017/CB09780511777110](https://doi.org/10.1017/CB09780511777110).
- [vG20] Rob van Glabbeek. Reactive bisimulation semantics for a process algebra with time-outs, 2020. arXiv:[2008.11499](https://arxiv.org/abs/2008.11499).
- [vG21] Rob van Glabbeek. Failure trace semantics for a process algebra with time-outs. *Logical Methods in Computer Science*, 17(2):11:1–11:40, 2021. URL: <https://lmcs.episciences.org/7398>, doi:[10.23638/LMCS-17\(2:11\)2021](https://doi.org/10.23638/LMCS-17(2:11)2021).
- [Wen21a] Makarius Wenzel. The Isabelle system manual, February 2021. Accessed on 28.05.2021. URL: <https://isabelle.in.tum.de/doc/system.pdf>.
- [Wen21b] Makarius Wenzel. The Isabelle/Isar reference manual, February 2021. Accessed on 28.05.2021. URL: <https://isabelle.in.tum.de/doc/isar-ref.pdf>.
- [WEP<sup>+</sup>16] Tjark Weber, Lars-Henrik Eriksson, Joachim Parrow, Johannes Borgström, and Ramunas Gutkovas. Modal logics for nominal transition systems. *Archive of Formal Proofs*, 2016. URL: [https://isa-afp.org/entries/Modal\\_Logics\\_for\\_NTS.html](https://isa-afp.org/entries/Modal_Logics_for_NTS.html).



# Appendix A

## Introduction to Isabelle

Isabelle is an interactive proof assistant and Isabelle/HOL is an implementation of *higher-order logic* in Isabelle. With it, one can interactively prove propositions about theories that are formalised in terms of higher-order logic. Many theories have been formalised (and many theorems proven) in Isabelle/HOL and are publicly available.<sup>1</sup>

In this appendix, I will give a short introduction into the most important concepts of Isabelle. For an extensive tutorial, see [Nip21]. A complete documentation can be found in [Wen21b].

### Simple Definitions

The command `definition` defines a term by establishing an equality, denoted by  $\equiv$ . This term can be a function or a constant (i.e. 0-ary function). Predicates are functions to Boolean values.

Definitions are annotated by their type. As an example, we define the predicate `even`, which maps an integer to a Boolean value.

```
definition even :: (int  $\Rightarrow$  bool)
  where (even n  $\equiv \exists$  m::int . n = 2 * m)
```

Functions can be defined in uncurried form (e.g. (int  $\times$  int)  $\Rightarrow$  bool) or in curried form (e.g. int  $\Rightarrow$  int  $\Rightarrow$  bool). As a very trivial example, we can define equality predicates for integers. Compared to the curried version, the uncurried version does not allow for easy pattern matching. This is why, in this thesis, I usually specify functions in curried form.

```
definition equal_uncurried :: ((int  $\times$  int)  $\Rightarrow$  bool)
  where (equal_uncurried pair  $\equiv \exists$  n m. pair = (n, m)  $\wedge$  n = m)
```

```
definition equal_curried :: (int  $\Rightarrow$  int  $\Rightarrow$  bool)
  where (equal_curried n m  $\equiv$  n = m)
```

---

<sup>1</sup>see Isabelle’s Archive of Formal Proofs at [isa-afp.org](http://isa-afp.org)

We can also use type variables (prefixed with an apostrophe, e.g. 'a) instead of concrete types to get more abstract terms.

```
definition equal_abstract :: ('a  $\Rightarrow$  'a  $\Rightarrow$  bool)
  where (equal_abstract a b  $\equiv$  a = b)
```

For a less trivial example, we define a predicate `symmetric` that determines whether a given relation is symmetric. An arbitrary homogeneous relation in curried form has the type 'a  $\Rightarrow$  'a  $\Rightarrow$  bool.

```
definition symmetric :: (('a  $\Rightarrow$  'a  $\Rightarrow$  bool)  $\Rightarrow$  bool)
  where (symmetric R  $\equiv$   $\forall$  a b. R a b  $\longrightarrow$  R b a)
```

We can also assign notation to a term during the definition, where `_` is a placeholder (and the numbers behind the notation specification represent priorities for parsing, which may be ignored by the reader).

```
definition approx :: (int  $\Rightarrow$  int  $\Rightarrow$  bool)
  ((_  $\approx$  _) [50, 50] 50)
  where (n  $\approx$  m  $\equiv$  n=m-1  $\vee$  n=m  $\vee$  n=m+1)
```

Abbreviations are used the same way as definitions, except that, in order to use the equality established by definitions in proofs, we need to explicitly refer to the definition, whereas abbreviations are always expanded internally by the proof system. An example a little further down below should clarify the distinction.

```
abbreviation reflexive :: (('a  $\Rightarrow$  'a  $\Rightarrow$  bool)  $\Rightarrow$  bool)
  where (reflexive R  $\equiv$   $\forall$  a. R a a)
```

## Proving Propositions

Propositions can be stated using any of the commands `proposition`, `lemma`, `theorem`, `corollary`, and require a proof.

Since Isabelle is an *interactive* proof assistant, proofs are usually meant to be spelled out in code so as to be readable by humans, and the validity of individual steps is verified by certain automated proof methods (e.g. `simp`, `arith`, `auto`, `fast`, `blast`, ...).

As an example, we will show that the relation `approx` is `symmetric`.

Since `symmetric` was defined using the command `definition`, we need to explicitly unfold it, where `symmetric_def` is the fact (about the equality) introduced by the definition.

The method specified after the command `proof` adjusts the proof goal in some way. Ideally, the proof steps should be clear to the reader even without seeing what exactly the automated methods are doing. I have explained each of the steps using comments below.

```

proposition ⟨symmetric approx⟩
  unfolding symmetric_def
proof (clarify)
  — We want to show that for any  $n$  and  $m$  with  $n \approx m$ , we have  $m \approx n$ .
  fix  $n\ m$ 
  assume  $\langle n \approx m \rangle$ 
  — Using the definition of approx, we know this about  $n$  and  $m$ .
  hence  $\langle n=m-1 \vee n=m \vee n=m+1 \rangle$  unfolding approx_def .
  thus  $\langle m \approx n \rangle$ 
  — With disjunction elimination, we examine each case in a sub-proof.
  proof (elim disjE)
    assume  $\langle n = m - 1 \rangle$ 
    hence  $\langle m = n + 1 \rangle$  by arith
    thus  $\langle m \approx n \rangle$  unfolding approx_def by blast
  next
    assume  $\langle n = m \rangle$ 
    thus  $\langle m \approx n \rangle$  using approx_def by blast
  next
    assume  $\langle n = m + 1 \rangle$ 
    hence  $\langle m = n - 1 \rangle$  by arith
    thus  $\langle m \approx n \rangle$  using approx_def by blast
  qed
qed

```

This proof was probably more detailed than was necessary. By unfolding the other definition as well, this proposition can be proven directly with the proof method **arith**.

```

proposition ⟨symmetric approx⟩
  unfolding symmetric_def approx_def by arith

```

To see the difference between definitions and abbreviations, note that the following proposition is provable without unfolding **reflexive\_def** (since **reflexive** is an abbreviation, there is no such fact in this context).

```

proposition ⟨reflexive approx⟩
  unfolding approx_def by auto

```

In practice, of course, one has to strike a balance between transparency/comprehensibility and conciseness of proofs.

## Inductive Definitions

Inductively defined predicates can be given using premise-conclusion pairs and multiple clauses.

```

inductive even_ind :: ⟨int  $\Rightarrow$  bool⟩
  where
    ⟨even_ind 0⟩
    | ⟨even_ind  $n \implies$  even_ind ( $n+2$ )⟩

```

### Function Definitions

The command `function` also establishes equalities, but usually in more complex ways, so that it may not be obvious whether a function is well-defined. Hence, the well-definedness needs to be proved explicitly. (These proofs are often solved by the automated proof methods.)

The function is then assumed to be partial. The command `termination` introduces proof obligations to show that the function always terminates (and is thus total). For the example below, this is again proved automatically.

After proving well-definedness and totality, we have access to facts about the function that can be used in proofs, e.g. induction principles. More details can be found in [section 2.3](#), where we define a non-trivial function.

```
function factorial :: (nat  $\Rightarrow$  nat)
  where
    (n = 0  $\implies$  factorial n = 1)
  | (n > 0  $\implies$  factorial n = n * factorial (n-1))
  by auto
```

```
termination factorial using termination by force
```

### Data Types

With the command `datatype`, new types can be defined, possibly in dependence on existing types, by defining a set of (object) constructor functions. For example, we can (re-)define the type of natural numbers.

```
datatype natural_number =
  Zero — 0-ary base constructor
| Suc (natural_number) — unary recursive/inductive constructor
```

We can define type constructors, i.e. types depending on other types (to be distinguished from the object constructors above) by parameterising the type with type variables.

```
datatype ('a)list =
  Empty
| Cons ('a) (('a)list)
```

### Locales

Locales define a context consisting of type variables, object variables, and assumptions. These can be accessed in the entire context. Locales can be instantiated by specifying concrete types/objects (or variables from another context) for the type/object variables, and extended to form new locales. We can reenter the locale contexts later on, using the command `context`. [Section 2.1](#) provides a good example for how locales are used in Isabelle to formalise linear transition systems.

## Appendix B

# Infinitary Hennessy-Milner Logic

We will show that a modal characterisation of strong bisimilarity is possible without any assumptions about the cardinality of derivative sets  $\text{Der}(\mathbf{p}, \alpha)$ , using infinitary HML (with conjunction of arbitrary cardinality).

Instead of formalising formulas under a conjunction as a countable set,<sup>1</sup> we use an index set of arbitrary type  $\mathbf{I} :: 'x \text{ set}$  and a mapping  $\mathbf{F} :: 'x \Rightarrow ('a, 'x)\text{HML\_formula}$  so that each element of  $\mathbf{I}$  is mapped to a formula. This closely resembles the semantics of  $\bigwedge_{i \in I} \varphi_i$ . Instead of using partial mappings  $\mathbf{F} :: 'x \Rightarrow ('a, 'x)\text{HML\_formula option}$ , I included a constructor `HML_true` and implicitly assume that  $\mathbf{F}$  maps to `HML_true` for all objects of type  $'x$  that are not elements of  $\mathbf{I}$ .

```
datatype ('a, 'x)HML_formula =  
  HML_true  
| HML_conj <'x set> <'x  $\Rightarrow$  ('a, 'x)HML_formula>  
| HML_neg <('a, 'x)HML_formula>  
| HML_poss <'a> <('a, 'x)HML_formula>
```

### Satisfaction Relation

Data types cannot be used with variable parameter types  $'x$  in concrete contexts; so when using our data type in the context `lts`, we use the type of processes `'s` as the type for conjunction index sets.

---

<sup>1</sup>Note that it is not possible to define a type with a constructor depending on a set of the type itself, i.e. `HML_conj <('a)HML_formula set>` would not yield a valid type.

Since this suffices to proof the modal characterisation, we can conclude that it suffices for the cardinality of conjunction to be equal to the cardinality of the set of processes  $Proc$ . As we can deduce from the part of the proof where formula conjunction is used, a weaker requirement would be to allow for conjunction of cardinality equal to  $\max_{\substack{p \in Proc \\ \alpha \in Act}} |Der(p, \alpha)|$ .

The remainder of this appendix follows the same structure as that in [section 2.3](#). The explanations from there mostly apply here as well.

**context** `lts begin`

```

function satisfies :: ⟨'s ⇒ ('a, 's) HML_formula ⇒ bool⟩
  (⟨_ | _⟩ [50, 50] 50)
  where
    ⟨(p | HML_true) = True⟩
  | ⟨(p | HML_conj I F) = (∀ i ∈ I. p | (F i))⟩
  | ⟨(p | HML_neg φ) = (¬ p | φ)⟩
  | ⟨(p | HML_poss α φ) = (∃ p'. p ↦α p' ∧ p' | φ)⟩
  ⟨proof⟩

inductive_set HML_wf_rel :: ⟨('s × ('a, 's) HML_formula) rel⟩
  where
    ⟨φ = F i ∧ i ∈ I ⇒ ((p, φ), (p, HML_conj I F)) ∈ HML_wf_rel⟩
  | ⟨((p, φ), (p, HML_neg φ)) ∈ HML_wf_rel⟩
  | ⟨((p, φ), (p', HML_poss α φ)) ∈ HML_wf_rel⟩

lemma HML_wf_rel_is_wf: ⟨wf HML_wf_rel⟩
  ⟨proof⟩

termination satisfies using HML_wf_rel_is_wf
  by (standard, (simp add: HML_wf_rel.intros)+)

```

## Modal Characterisation of Strong Bisimilarity

```

definition HML_equivalent :: ('s  $\Rightarrow$  's  $\Rightarrow$  bool)
  where (HML_equivalent p q
     $\equiv (\forall \varphi::('a, 's) \text{HML\_formula. } (p \models \varphi) \longleftrightarrow (q \models \varphi))$ )

```

```

lemma distinguishing_formula:
  assumes  $\neg \text{HML\_equivalent } p \ q$ 
  shows  $\exists \varphi. p \models \varphi \wedge \neg q \models \varphi$ 
  <proof>

```

```

lemma HML_equivalent_symm:
  assumes  $\text{HML\_equivalent } p \ q$ 
  shows  $\text{HML\_equivalent } q \ p$ 
  <proof>

```

```

lemma strong_bisimilarity_implies_HML_equivalent:
  assumes  $p \leftrightarrow q$   $p \models \varphi$ 
  shows  $q \models \varphi$ 
  using assms
proof (induct  $\varphi$  arbitrary: p q)
  case HML_true
  then show ?case
    by force
next
  case (HML_conj X F)
  then show ?case
    by force
next
  case (HML_neg  $\varphi$ )
  then show ?case
    using satisfies.simps(3) strongly_bisimilar_symm
    by meson
next
  case (HML_poss  $\alpha \varphi$ )
  then show ?case
    using satisfies.simps(4) strongly_bisimilar_step(1)
    by meson
qed

```

```

lemma HML_equivalence_is_SB:
  shows ⟨SB HML_equivalent⟩
proof -
  {
    fix p q p' α
    assume ⟨HML_equivalent p q⟩ ⟨p ⟶α p'⟩
    assume ⟨∀ q' ∈ Der(q, α). ¬ HML_equivalent p' q'⟩

    hence exists_φq': ⟨∀ q' ∈ Der(q, α). ∃ φ. p' ⊨ φ ∧ ¬ q' ⊨ φ⟩
      using distinguishing_formula by blast

    let ?I = ⟨Der(q, α)⟩
    let ?F = ⟨(λ q'. SOME φ. p' ⊨ φ ∧ ¬ q' ⊨ φ)⟩
    let ?φ = ⟨HML_conj ?I ?F⟩

    from exists_φq' have ⟨p' ⊨ ?φ⟩
      by (smt (z3) satisfies.simps(2) someI_ex)
    hence ⟨p ⊨ HML_poss α ?φ⟩ using ⟨p ⟶α p'⟩ by auto

    from exists_φq' have ⟨∀ q' ∈ Der(q, α). ¬ q' ⊨ ?φ⟩
      by (smt (z3) satisfies.simps(2) someI_ex)
    hence ⟨¬ q ⊨ HML_poss α ?φ⟩ by simp

    from ⟨p ⊨ HML_poss α ?φ⟩ ⟨¬ q ⊨ HML_poss α ?φ⟩ have False
      using ⟨HML_equivalent p q⟩ HML_equivalent_def by blast
  }

  thus ⟨SB HML_equivalent⟩ unfolding SB_def
    using HML_equivalent_symm by blast
qed

theorem modal_characterisation_of_strong_bisimilarity:
  shows ⟨p ↔ q ⟷ (∀ φ. p ⊨ φ ⟷ q ⊨ φ)⟩
proof
  show ⟨p ↔ q ⟹ ∀ φ. (p ⊨ φ) = (q ⊨ φ)⟩
    using strong_bisimilarity_implies_HML_equivalent
      strongly_bisimilar_symm
    by blast
next
  show ⟨∀ φ. (p ⊨ φ) = (q ⊨ φ) ⟹ p ↔ q⟩
    using HML_equivalence_is_SB HML_equivalent_def
      strongly_bisimilar_def
    by blast
qed

end — of context lts

```



## Appendix C

# Example Instantiation

To complete the proofs from [chapter 3](#), I will show that mappings `stabilise` ( $\varepsilon(\_)$ ) and `in_env` ( $\vartheta?[_](\_)$ ), whose existence we assumed up to now, do, in fact, exist. I will define example mappings and show that, together with these, arbitrary `lts_timeout` can be interpreted as `lts_timeout_mappable`, thereby showing that the reductions are valid for arbitrary  $\text{LTS}_t$ s.

First, we define the types for  $\text{Proc}_\vartheta$  and  $\text{Act}_\vartheta$  in dependence on arbitrary types `'s` and `'a` for  $\text{Proc}$  and  $\text{Act}$ , respectively:

```
datatype ('s,'a)Proc_ϑ = triggered 's | stable ⟨'a set⟩ 's | DumpState
datatype ('a)Act_ϑ = act 'a | t_ε | ε ⟨'a set⟩ | DumpAction
```

Since  $\text{Act} \neq \text{Act}_\vartheta$ , we define a new predicate `tran_mappable`.

```
context lts_timeout begin

inductive tran_mappable
  :: ⟨'s  $\Rightarrow$  ('a)Act_ϑ  $\Rightarrow$  's  $\Rightarrow$  bool⟩
  where ⟨tran p α p'  $\Rightarrow$  tran_mappable p (act α) p'⟩
```

We can now specify mappings `stabilise` and `in_env`, mapping those  $X$  for which  $\varepsilon_X$  and  $\vartheta_X$  are undefined to the `DumpAction/DumpState`.

```
function stabilise :: ⟨('a)Act_ϑ set  $\Rightarrow$  ('a)Act_ϑ⟩
  where
    ⟨ $\forall \alpha \in X. (\exists \alpha'. \alpha = \text{act } \alpha') \Rightarrow \text{stabilise } X = \varepsilon \{ \alpha' \mid \text{act } \alpha' \in X \}$ ⟩
    | ⟨ $\exists \alpha \in X. (\nexists \alpha'. \alpha = \text{act } \alpha') \Rightarrow \text{stabilise } X = \text{DumpAction}$ ⟩
    ⟨proof⟩
termination ⟨proof⟩
```

```

function in_env :: ⟨('a)Act_θ set option ⇒ 's ⇒ ('s,'a)Proc_θ⟩
  where
    ⟨in_env None p = triggered p⟩
  | ⟨∀ α∈X. (∃ α'. α = act α') ⇒
      in_env (Some X) p = stable {α' . act α' ∈ X} p⟩
  | ⟨∃ α∈X. (∄ α'. α = act α') ⇒
      in_env (Some X) p = DumpState⟩
  ⟨proof⟩
termination ⟨proof⟩

```

We show that, with these mappings, any `lts_timeout` (the context we are in) is an `lts_timeout_mappable`: when the variables that were fixed in the locale definition are instantiated by the terms and mappings from the current context, we prove that the assumptions of the locale definition hold. Thus, the reduction works for all  $LTS_t$ s.


```

lemma is_mappable: ⟨lts_timeout_mappable
  tran_mappable (act τ) (act t) t_ε stabilise in_env⟩
  ⟨proof⟩

```

end — of context `lts_timeout`

### A Tiny Example $LTS_t$



```

datatype Proc = p0|p1|p2|q0|q1
datatype Act = τ|t
inductive Tran :: ⟨Proc ⇒ Act ⇒ Proc ⇒ bool⟩
  where
    ⟨Tran p0 τ p1⟩
  | ⟨Tran p0 t p2⟩
  | ⟨Tran q0 τ q1⟩

```

We interpret the `Tran` predicate as an `lts_timeout`, and then together with our mappings as an `lts_timeout_mappable`.

```

interpretation tiny_lts:
  lts_timeout Tran τ t
  by (simp add: lts_timeout.intro)
interpretation tiny_lts_mappable:
  lts_timeout_mappable tiny_lts.tran_mappable ⟨act τ⟩ ⟨act t⟩ ⟨t_ε⟩
  tiny_lts.stabilise tiny_lts.in_env
  using tiny_lts.is_mappable .
— (notation assignments omitted from thesis document)

```

We can now prove a few lemmas about our example  $\text{LTS}_t$  that we would need for any bisimilarity proofs. I abstained from actually including a bisimilarity proof, but these lemmas should, hopefully, suffice to convince you that it would be possible.

```

lemma ⟨tiny_lts_mappable.visible_actions = ∅⟩
proof -
  have ⟨tiny_lts.visible_actions = ∅⟩
    using tiny_lts.visible_actions_def
    Act.exhaust Collect_cong empty_def
    by auto
  moreover have ⟨tiny_lts_mappable.visible_actions
    = image (λ α. act α) tiny_lts.visible_actions⟩
    using Act.exhaust
    tiny_lts.tran_mappable.simps
    tiny_lts.visible_actions_def
    tiny_lts_mappable.visible_actions_def
    by auto
  ultimately show ?thesis by force
qed

lemma  $\nexists P'. \vartheta[\emptyset](p0) \mapsto^\vartheta(\text{act } t) P'$ 
proof safe
  fix P'
  assume ⟨ $\vartheta[\emptyset](p0) \mapsto^\vartheta(\text{act } t) P'$ ⟩
  hence ⟨idle p0 ∅⟩
    using tiny_lts_mappable.generation_sys_timeout by blast

  have ⟨p0  $\mapsto$ (act  $\tau$ ) p1⟩ using Tran.intros(1)
    by (simp add: tiny_lts.tran_mappable.intros)

  with ⟨idle p0 ∅⟩ show False
    unfolding tiny_lts_mappable.initial_actions_def by blast
qed

```



# German-language summary / Zusammenfassung in Deutscher Sprache

## Reduktion von Starker Reaktiver Bisimilarität auf Starke Bisimilarität

In dieser Arbeit zeige ich, dass es möglich ist, die Bestimmung von starker reaktive Bisimilarität (strong reactive bisimilarity), wie sie von Rob van Glabbeek in [vG20] eingeführt wurde, auf die Bestimmung von gewöhnlicher starker Bisimilarität zu reduzieren, indem ich ein Mapping spezifiziere, welches ein Modell des geschlossenen Systems, bestehend aus einem zugrundeliegenden reaktiven System und dessen Umgebung, liefert. Ich habe alle Konzepte, die ich in dieser Arbeit diskutiere, sowie alle Beweise, die ich durchgeführt habe, im interaktiven Beweisassistenten Isabelle formalisiert.

Reaktive Systeme sind Systeme, die kontinuierlich mit ihrer Umgebung (z. B. einem Benutzer) interagieren und deren Verhalten weitgehend von dieser Interaktion abhängig ist [HP85]. Sie können mit Hilfe von beschrifteten Übergangssystemen (labelled transition systems, LTSs) modelliert werden [Kel76]; grob gesagt ist ein LTS ein beschrifteter gerichteter Graph, dessen Knoten den Zuständen eines reaktiven Systems und dessen Kanten den Übergängen zwischen diesen Zuständen entsprechen.

Ein Benutzer, der mit einem System interagiert, kann es nur in Bezug auf die Interaktionen wahrnehmen, auf die das System reagiert, d. h. der interne Zustand des Systems ist dem Benutzer verborgen. Daraus ergibt sich der Begriff der Verhaltens-/Beobachteräquivalenz: Zwei nicht identische Systeme können, aus Sicht des Benutzers, ein äquivalentes Verhalten vorzeigen. Die einfachste derartige Äquivalenz ist als *starke Bisimilarität* bekannt.

In klassischen LTSs kann ein System nicht auf die Abwesenheit von Interaktion reagieren, da angenommen wird, dass es einfach auf irgendeine Interaktion wartet. Intuitiv kann ein System jedoch mit einer Uhr ausgestattet sein und eine Aktivität ausführen, wenn es eine bestimmte Zeit lang keine Interaktion des Benutzers gesehen hat. Ein solches System wäre mit

klassischer LTS-Semantik nicht beschreibbar.

In [vG21] führt Rob van Glabbeek beschriftete Übergangssysteme mit Timeouts ( $LTS_t$ ) ein, mit denen auch solche Systeme modelliert werden können. Die zugehörige Äquivalenz ist in [vG20] als *starke reaktive Bisimilarität* (strong reactive bisimilarity) gegeben.

Als erstes Hauptergebnis dieser Arbeit zeige ich, dass es möglich ist, die Bestimmung starker reaktiver Bisimilarität auf die Bestimmung starker Bisimilarität zu reduzieren.

Die Strategie zur Reduktion von reaktiver Bisimilarität auf starke Bisimilarität basiert auf der Tatsache, dass das Konzept der starken reaktiven Bisimilarität eine explizite Berücksichtigung der Umgebungen erfordert, in denen spezifizierte Systeme existieren können. Konkret spezifiziere ich ein Mapping von  $LTS_t$ s auf LTSs, wobei jeder Zustand des gemappten LTS einem Zustand des ursprünglichen  $LTS_t$  in einer bestimmten Umgebung entspricht.

Eine weitere interessante Möglichkeit, das Verhalten eines LTS zu untersuchen, ist die Verwendung von modalen Logiken, bei denen Formeln bestimmte Eigenschaften beschreiben und auf Zuständen eines LTS ausgewertet werden. Die in der Forschung zu reaktiven Systemen am häufigsten verwendete Modallogik ist als Hennessy-Milner-Logik (HML) bekannt. Eine Erweiterung von HML für die Auswertung auf Zuständen einer  $LTS_t$  ist ebenfalls in [vG20] gegeben; ich bezeichne diese Erweiterung als Hennessy-Milner-Logik mit Zeitüberschreitungen ( $HML_t$ ).

Als zweites Hauptergebnis dieser Arbeit zeige ich, dass es möglich ist, die Erfüllung von  $HML_t$ -Formeln in  $LTS_t$ s auf die Erfüllung von HML in LTSs zu reduzieren (unter Verwendung eines weiteren Mappings für Formeln, zusammen mit dem Mapping aus der ersten Reduktion).

# Statement of Authorship

I hereby declare that the thesis submitted is my own, unaided work, completed without any unpermitted external help. Only the sources and resources listed were used.

The independent and unaided completion of the thesis is affirmed by affidavit:

Berlin, \_\_\_\_\_

\_\_\_\_\_

Signature